code:
;クンニ

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME2
LOCAL:0 = IS_RIDE(MPLY:0, MTAR:0)

IF MTAR:0 == -1
	LOCALS:0 = Cunnilingus
ELSE
	;男性プレイヤーがターゲットに顔面騎乗中
	IF LOCAL:0 && !HAS_VAGINA(MPLY:0)
		LOCALS:0 = Facing Cunnilingus
		;非男性プレイヤーがターゲットに顔面騎乗中 or ターゲットからプレイヤーにフェラ・クンニ中
	ELSEIF LOCAL:0 || SEARCH_EQUIP(11, MTAR:0, MPLY:0) >= 0 || SEARCH_EQUIP(2, MTAR:0, MPLY:0) >= 0
		IF HAS_PENIS(MTAR:0) && HAS_VAGINA(MTAR:0)
			LOCALS:0 = 69(C)
		ELSE
			LOCALS:0 = 69
		ENDIF
	ELSE
		LOCALS:0 = Cunnilingus
	ENDIF
ENDIF

RESULTS:0 = Do %LOCALS:0%
RESULTS:1 = Get ordered to %LOCALS:0%
RESULTS:2 = Order %LOCALS:0%
RESULTS:3 = Get %LOCALS:0%
RESULTS:4 = Make do %LOCALS:0%
RESULTS:5 = Show %LOCALS:0% off

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE2
;共通部分
CALL COM_ABLE_COMMON(2)
SIF RESULT == 0
	RETURN 0
;プレイヤーは最大で1人まで
SIF MPLY_NUM <= 0 || MPLY_NUM > 1
	RETURN 0
;ターゲットは最大で1人まで
SIF MTAR_NUM <= 0 || MTAR_NUM > 1
	RETURN 0
;対象にＶが必要
SIF !HAS_VAGINA(MTAR:0)
	RETURN 0
SIF !CAN_LICK_GROIN(MPLY:0, MTAR:0)
	RETURN 0
;target is anal facesitting player
SIF SEARCH_EQUIP(102, MTAR:0, MPLY:0) >= 0
	RETURN 0
;Although this is technically a Ｃ action, players won't understand if cunnilingus is blocked by the clit cap or similar.
;SIF IS_EQUIP_TARGET(MTAR:0, 63)
;	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM2
;実行判定
CALL COM_ORDER_COMMON
SIF RESULT == 0
	RETURN 0

;顔面騎乗中は強制的に非継続にする
IF IS_RIDDEN(MPLY:0)
	TFLAG:非継続フラグ = 1
ENDIF

;●プレイヤーについて処理
DOWNBASE:(MPLY:0):体力 += 120

EXP:(MPLY:0):性技経験値 += 1

SOURCE:(MPLY:0):奉仕 = SERVE_HOUSHI(MPLY:0, 300)
SOURCE:(MPLY:0):接触 = 50
SOURCE:(MPLY:0):性行動 = 300

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 150, 100)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MPLY:0):奉仕経験取得可能 = 1

;膣口へのキス
CALL KISS_COMMON(MPLY:0, @"%ANAME(MTAR:0)%の秘貝", GET_SITUATION_BY_TRAIN_MODE())

;●ターゲットについて処理
DOWNBASE:(MTAR:0):体力 += 60

SOURCE:(MTAR:0):液体追加 = 300
SOURCE:(MTAR:0):露出 = 150
SOURCE:(MTAR:0):逸脱 = 50
SOURCE:(MTAR:0):接触 = 50
SOURCE:(MTAR:0):快Ｃ = SENSE_HOUSHI(MPLY:0, MTAR:0, 1500)
SOURCE:(MTAR:0):性行動 = 300

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 150, 0)

IF TALENT:(MPLY:0):舌使い
	TIMES SOURCE:(MTAR:0):快Ｃ, 1.50
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = 0

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 3

CALL FL_PLAYSE("Lick_1.mp3",,"SEX")

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP2
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP2(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;●プレイヤーの処理
DOWNBASE:(LOCAL:2):体力 += 20

EXP:(LOCAL:2):性技経験値 += 1

SOURCE:(LOCAL:2):奉仕 += SERVE_HOUSHI(LOCAL:2, 100)
SOURCE:(LOCAL:2):接触 += 25
SOURCE:(LOCAL:2):性行動 += 100

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:2):奉仕経験取得可能 = 1

;●ターゲットの処理
DOWNBASE:(LOCAL:3):体力 += 10

SOURCE:(LOCAL:3):液体追加 += 150
SOURCE:(LOCAL:3):露出 += 75
SOURCE:(LOCAL:3):逸脱 += 25
SOURCE:(LOCAL:3):接触 += 25
SOURCE:(LOCAL:3):性行動 += 100

IF TALENT:(LOCAL:2):舌使い
	SOURCE:(LOCAL:3):快Ｃ += SENSE_HOUSHI(LOCAL:2, LOCAL:3, 1125)
ELSE
	SOURCE:(LOCAL:3):快Ｃ += SENSE_HOUSHI(LOCAL:2, LOCAL:3, 750)
ENDIF

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE2(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%にクンニ中

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP2(ARG:0)
PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%の秘唇を舌で舐めている…

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP2(ARG:0)

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER2(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 75

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 4)

LOCAL:0 = GET_PALAMLV(PALAM:(ARG:0):欲情)
CALL COM_ORDER_ELEMENT(ARG:0, @"欲情Lv{LOCAL:0}", MIN(LOCAL:0 * 2, 20))

IF TCVAR:(ARG:0):媚薬
	CALL COM_ORDER_ELEMENT(ARG:0, "媚薬", 6)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -1)
ENDIF
IF TALENT:(ARG:0):汚臭鈍感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭鈍感", 1)
ENDIF
IF TALENT:(ARG:0):汚臭敏感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭敏感", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):快感の否定
	CALL COM_ORDER_ELEMENT(ARG:0, "快感の否定", -1)
ENDIF
IF TALENT:(ARG:0):女嫌い
	CALL COM_ORDER_ELEMENT(ARG:0, "女嫌い", -7)
ENDIF

;合意
IF !TALENT:(ARG:0):合意 && !TALENT:(ARG:0):親友
	CALL COM_ORDER_ELEMENT(ARG:0, "合意なし", -10)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE2
#DIM  DYNAMIC 被覆衣装部位
#DIM  DYNAMIC 被覆衣装ID
#DIMS DYNAMIC 被覆衣装描写

;下着で秘部が覆われているか確認する
CALL IS_COVERED(MTAR:0, 被覆部位_秘部, 1)
被覆衣装部位 = RESULT:0
被覆衣装ID = RESULT:1
IF RESULT:1 != 0 && (被覆衣装部位 == 衣装部位_下着下 || 衣装部位_上下下着)
	TRYCCALLFORM CLOTH_DESC_%CLOTH_PART_NAME_ENG:被覆衣装部位%_{被覆衣装ID}
		被覆衣装描写 '= RESULTS
	CATCH
		TRYCCALLFORM CLOTH_NAME_%CLOTH_PART_NAME_ENG:被覆衣装部位%_{被覆衣装ID}
			被覆衣装描写 '= RESULTS
		CATCH
		ENDCATCH
	ENDCATCH
ELSE
	被覆衣装描写 '= ""
ENDIF

LOCAL:5 = SEARCH_EQUIP(101, MPLY:0, MTAR:0)
LOCAL:6 = SEARCH_EQUIP(102, MPLY:0, MTAR:0)
LOCAL:7 = SEARCH_EQUIP(11, MTAR:0, MPLY:0)
LOCAL:8 = SEARCH_EQUIP(2, MTAR:0, MPLY:0)
LOCAL:9 = SEARCH_EQUIP(101, MTAR:0, MPLY:0)
LOCALS:1 =
IF HAS_PENIS(MTAR:0)
	LOCALS:1 = 竿の付け根にある
ELSEIF PALAM:(MTAR:0):潤滑 >= PALAMLV:5
	LOCALS:1 = 濡れそぼった
ELSEIF TALENT:(MTAR:0):体格 == 体格_子供
	LOCALS:1 = 幼い
ENDIF

IF IS_PIERCED(MTAR:0, ピアス_ラビア)
	LOCALS:2 = ピアスの取り付けられた
ELSE
	LOCALS:2 =
ENDIF

SELECTCASE GET_INMOU_LEVEL(MTAR:0)
	CASE 1, 2
		LOCALS:3 = 陰毛の生い茂る
	CASEELSE
		LOCALS:3 =
ENDSELECT

IF CUM_GET_AMOUNT:(MTAR:0):射精部位_膣内 > 0
	LOCALS:4 = 白濁をこぼしている
ELSE
	LOCALS:4 =
ENDIF

;プレイヤーがターゲットに顔面騎乗中かつ男の場合
IF LOCAL:5 >= 0 && !HAS_VAGINA(MPLY:0)
	SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
		CASE 0
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に自身の玉を舐めさせながら、%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めほぐし%TARGET_STR("た", "てきた")%…
			;ターゲットに主導権
		CASE 1
			PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の玉を舐めながら、%ANAME(MPLY:0)%に自身の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めさせた…
			;第三者に主導権
		CASEELSE
			PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に%ANAME(MTAR:0)%の%LOCALS:2%秘唇を舐めるように命令した
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に自身の玉を舐めさせながら、%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めほぐした…
	ENDSELECT
	;顔面騎乗の地の文をカット
	IF LOCAL:5 >= 0
		MEQUIP_CUT:(LOCAL:5) = 1
	ENDIF

	;非男性プレイヤーがターゲットに顔面騎乗中 or ターゲットからプレイヤーにフェラ・クンニ中
ELSEIF LOCAL:5 >= 0 || LOCAL:6 >= 0 || LOCAL:7 >= 0 || LOCAL:8 >= 0
	IF LOCAL:6 >= 0
		LOCALS:0 = アナル
	ELSEIF LOCAL:7 >= 0
		LOCALS:0 = ペニス
	ELSE
		LOCALS:0 = 秘唇
	ENDIF
	SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
		CASE 0
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に自身の%LOCALS:0%を舐めさせながら、%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めほぐし%TARGET_STR("た", "てきた")%…
			;ターゲットに主導権
		CASE 1
			PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の%LOCALS:0%を舐めながら、%ANAME(MPLY:0)%に自身の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めさせた…
			;第三者に主導権
		CASEELSE
			PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%と%ANAME(MTAR:0)%にシックスナインの体勢で絡み合うように命令した
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に自身の%LOCALS:0%を舐めさせながら、%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舐めほぐした…
	ENDSELECT
	;顔面騎乗・フェラ・クンニの地の文をカット
	IF LOCAL:5 >= 0
		MEQUIP_CUT:(LOCAL:5) = 1
	ENDIF
	IF LOCAL:6 >= 0
		MEQUIP_CUT:(LOCAL:6) = 1
	ENDIF
	IF LOCAL:7 >= 0
		MEQUIP_CUT:(LOCAL:7) = 1
	ENDIF
	IF LOCAL:8 >= 0
		MEQUIP_CUT:(LOCAL:8) = 1
	ENDIF

	;ターゲットからプレイヤーに顔面騎乗中
	;ELSEIF LOCAL:8 >= 0
	;	SELECTCASE GET_COM_INITIATIVE()
	;		;プレイヤーに主導権
	;		CASE 0
	;			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を顔の上に跨がらせたまま、下から%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:4%ワレメを貪るように舐め上げ%TARGET_STR("た", "てきた")%…
	;		;ターゲットに主導権
	;		CASE 1
	;			PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の顔に跨ったまま、グリグリと腰を押し付けて自身の%LOCALS:2%%LOCALS:3%%LOCALS:4%ワレメを舐めさせた…
	;		;第三者に主導権
	;		CASEELSE
	;			PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に%ANAME(MTAR:0)%の秘唇を舐めるように命令した
	;			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を顔の上に跨がらせたまま、下から%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:4%ワレメを舐め上げた…
	;	ENDSELECT
	;	;顔面騎乗の地の文をカットする
	;	MEQUIP_CUT:(LOCAL:8) = 1

	;プレイヤーが拘束中
ELSEIF IS_BIND(MPLY:0)
	SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
		CASE 0
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを顔に押し当てさせると、舌で激しく舐め回し%TARGET_STR("た", "てきた")%…
			;ターゲットに主導権
		CASE 1
			PRINTFORMW %ANAME(MTAR:0)%は身動きの取れない%ANAME(MPLY:0)%の口に%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを押し付け、舐めさせ奉仕させた…
			;第三者に主導権
		CASEELSE
			PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に対して、%ANAME(MPLY:0)%に秘唇を舐めさせるように命令した
			PRINTFORMW %ANAME(MTAR:0)%は身動きの取れない%ANAME(MPLY:0)%の口に%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを押し付け、舐めさせた…
	ENDSELECT

ELSE
	SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
		CASE 0
			IF 被覆衣装描写 == ""
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舌で舐め上げ%TARGET_STR("た", "てきた")%…
			ELSE
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%被覆衣装描写%%SPLIT_RAND("の上から/をずらし",1)%、%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舌で舐め上げ%TARGET_STR("た", "てきた")%…
			ENDIF
			;ターゲットに主導権
		CASE 1
			IF GETBIT(TALENT:(MTAR:0):淫乱系, 素質_淫乱_サド)
				LOCALS:4 = サディステックな笑みを浮かべながら、
			ELSEIF GETBIT(TALENT:(MTAR:0):淫乱系, 素質_淫乱_淫乱)
				LOCALS:4 = 恍惚とした表情を浮かべながら、
			ENDIF
			IF 被覆衣装描写 == ""
				PRINTFORMW %ANAME(MTAR:0)%は%LOCALS:4%%ANAME(MPLY:0)%の口に%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを押し付け、舐めさせ奉仕させた…
			ELSE
				PRINTFORMW %ANAME(MTAR:0)%は%LOCALS:4%%ANAME(MPLY:0)%の口に%被覆衣装描写%%SPLIT_RAND("の上から/越しに/から露になった、",1)%%LOCALS:2%%LOCALS:3%%LOCALS:1%ワレメを押し付け、舐めさせ奉仕させた…
			ENDIF
			;第三者に主導権
		CASEELSE
			PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に%ANAME(MTAR:0)%の秘唇を舐めるように命令した
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%LOCALS:2%%LOCALS:3%%LOCALS:1%%LOCALS:4%ワレメを舌で舐め上げた…
	ENDSELECT
ENDIF

IF CUM_CUR_AMOUNT:(MTAR:0):射精部位_膣内
	PRINTFORM %SPLIT_RAND("熱い/粘つく/濃厚な/新鮮な", 1)%精液が

	IF ABL:(MPLY:0):精愛 >= 4
		PRINTFORML %SPLIT_RAND("精液好きの/精愛味覚を得た/精液に飢えた", 1)%%ANAME(MPLY:0)%の口に滴ると、%PRONOUN(MPLY:0)%は悦びの呻き声を漏らした…
	ELSE
		PRINTFORML %ANAME(MPLY:0)%の口に滴るが、%PRONOUN(MPLY:0)%は奉仕を余儀なくされた…
	ENDIF
	;target eats 1/4th + 3 cum, up to 20
	LOCAL = LIMIT(MIN(((CUM_CUR_AMOUNT:(MTAR:0):射精部位_膣内 / 4) + 3), CUM_CUR_AMOUNT:(MTAR:0):射精部位_膣内) , 0, 20)
	CUM_CUR_AMOUNT:(MTAR:0):射精部位_膣内 -= LOCAL
	CUM_CUR_AMOUNT:(MPLY:0):射精部位_口 += LOCAL
	SIF CUM_CUR_AMOUNT:(MTAR:0):射精部位_膣内 == 0
		PRINTFORML %ANAME(MTAR:0)%の%STR_BODY("膣", MPLY:0)%の精液は全て%ANAME(MPLY:0)%が飲み干した
	;Cumlove exp increases by amount of cum
	EXP:(MPLY:0):精愛経験値 += LOCAL
	;Swallowing exp
	SIF ABL:(MPLY:0):精愛 >= 4
		EXP:(MPLY:0):口内射精経験 ++
	WAIT
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST2

IF MTAR:0 != MASTER || CONFIG:441
	LOCAL:2 = CUP:(MTAR:0):快Ｃ

	IF TCVAR:(MTAR:0):失神フラグ
		;気絶中
		PRINTFORML %ANAME(MTAR:0)%は秘唇を舐め上げられるたびに、ビクビクと体を震わせている
	ELSEIF LOCAL:2 < 300
		PRINTFORML %ANAME(MTAR:0)%のそこはまだしっとりと閉じたままだ。快楽を感じるにはもう少しかかるかもしれない
	ELSEIF LOCAL:2 < 1000
		PRINTFORML %ANAME(MTAR:0)%はクンニの刺激にじっと耐えているものの、確実に感じてきているようだ
	ELSEIF LOCAL:2 < 3000
		PRINTFORM 敏感な豆を舐められると感じるのか、%ANAME(MTAR:0)%は
		IF TALENT:(MTAR:0):恥じらい
			PRINT 羞恥に頬を赤く染め手で自分の口を抑えて声が出るのをこらえている
		ELSEIF TALENT:(MTAR:0):解放 || TALENT:(MTAR:0):快感に素直
			PRINTL 熱い吐息を吐きながら快楽に身をゆだねている
		ELSEIF TALENT:(MTAR:0):抑圧 || TALENT:(MTAR:0):孤高 || TALENT:(MTAR:0):快感の否定
			PRINTL 快楽を否定するように首を振って耐えている
		ELSE
			PRINTL 秘所から愛液を垂らしつつわずかに小さく吐息を漏らした
		ENDIF
	ELSEIF LOCAL:2 < 6000
		PRINTFORML %ANAME(MTAR:0)%の小さな豆を舌で擦り上げてやると%ANAME(MTAR:0)%は可愛らしい声を漏らしながらがくがくと身体を揺らした
	ELSEIF LOCAL:2 < 10000
		PRINTFORML %ANAME(MTAR:0)%は明らかな快楽の声をあげ、無意識に%ANAME(MPLY:0)%の頭を手で押さえてさらにその部分を押し付けてきた
	ELSE
		IF TALENT:(MTAR:0):Ｃ敏感
			PRINTFORML %ANAME(MTAR:0)%は敏感なクリトリスへの責めに激しく身もだえし、あられも無い声を張り上げ続けた
		ELSE
			PRINTFORML %ANAME(MTAR:0)%はクリトリスへの刺激に激しい喘ぎ声を上げながら悶え続けた
		ENDIF
	ENDIF
ENDIF

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN2
RETURN コマンド_ウフフ

;--------------------
;性嗜好用関数
;--------------------
@COM_PREFERENCE_PLAYER_2()
RETURN POWER(2, 性的嗜好_愛撫したい)

@COM_PREFERENCE_TARGET_2()
RETURN POWER(2, 性的嗜好_愛撫されたい)

@COM_STACK_SPERM_MTAR_TO_MPLY_2()
RETURN 射精部位_顔


code:
;会話

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME300
RESULTS:0 = Talk
RESULTS:1 = Listen
IF FLAG:調教モード == 調教_居住区 && MTAR != -1
	IF C_COND("行動:買物中",, MTAR)
		RESULTS:0 = Negotiate
		RESULTS:1 = Negotiate
		RETURN
	ELSEIF C_COND("行動:買春中",,MTAR)
		;売買春キャラ同士に性的関係が成立する必要がある
		SIF !IS_SEXUAL(MPLY, MTAR)
			RESULTS:0 = Hide dick
			RESULTS:1 = Hide dick
			RETURN
		SELECTCASE COM_OPTION:300
			CASE 0
				COM_NAME '= "春をひさぐ(なんでも)"
			CASE 1
				COM_NAME '= "春をひさぐ(お触りなし)"
			CASE 2
				COM_NAME '= "春をひさぐ(愛撫まで)"
			CASE 3
				COM_NAME '= "春をひさぐ(手淫まで)"
			CASE 4
				COM_NAME '= "春をひさぐ(口淫まで)"
			CASE 5
				COM_NAME '= "春をひさぐ(本番あり)"
			CASE 6
				COM_NAME '= "春をひさぐ(ＳＭ)"
			CASE 7
				COM_NAME '= "春をひさぐ(乱交)"
			ENDSELECT
		RESULTS:0 = %COM_NAME%
		RESULTS:1 = %COM_NAME%
	ENDIF
ENDIF

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE300
;共通部分
CALL COM_ABLE_COMMON(300)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公がプレイヤーのとき限定
SIF MPLY:0 != MASTER
	RETURN 0
;ターゲットがいなければ不可
SIF MTAR:0 == -1 || TFLAG:参加者数 == 0
	RETURN 0
;ディープキス中は不可
IF IS_EQUIP(MPLY:0, 342) && FLAG:調教モード != 調教_居住区
	RETURN 0
	;居住区モードではプレイヤーまたはターゲットの口が塞がっているなら不可
ELSEIF IS_M_HOLD(MPLY:0) || IS_M_HOLD(MTAR:0)
	RETURN 0
ENDIF
;双方意識があること
SIF !(IS_PLAYABLE(MPLY) && IS_PLAYABLE(MTAR))
	RETURN 0

RETURN 1

;-------------------------------------------------
;オプションを持つコマンドの場合オプション数を指定
;-------------------------------------------------
@COM_HAS_OPTION300
IF C_COND("行動:買春中",,MTAR)
	RETURN 7
ELSE
	RETURN 0
ENDIF

;-------------------------------------------------
;オプションの実行可否判定
;-------------------------------------------------
@COM_OPTION_ABLE300(COMOPTION)
#DIM COMOPTION
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM300
#DIM  DYNAMIC IS_SHOPPER
#DIM  DYNAMIC SHOPPER_SEDUCTION_LV

LOCAL:0 = 0
LOCAL:1 = 0

;印象値がベースとなります
LOCAL:0 += REL_LIKE:(MTAR:0):(MPLY:0) / 20
LOCAL:1 += REL_HATE:(MTAR:0):(MPLY:0) / 20
;抑圧・孤高によるLOCAL:0の減衰削りました（極端な結果になるので）
SIF TALENT:(MTAR:0):抑圧
	LOCAL:1 += YOKUATSU_RATE(MTAR:0) / 3
SIF TALENT:(MTAR:0):孤高
	LOCAL:1 += KOKOU_RATE(MTAR:0) / 3
;好感度条件最良時LOCAL:0 = 25 LOCAL:1 = 5が大成功率ちょうど良い塩梅と思われます（今のところ）
;最悪時はLOCAL:0 = 0 LOCAL:1 = 80となりますが、これでも初見話しかけて紫や依姫に斬りかかられることは稀なようです（今のところ）
;ただし会話はほとんど通用しません　外交で嫌われたせいです　酒を飲ませましょう
LOCAL:0 = LIMIT(LOCAL:0, 0, 25)
LOCAL:1 = LIMIT(LOCAL:1, 5, 80)

;コマンドの成否をTFLAG:コマンド成否にセット
CALL JUDGE_COM_RESULT(MTAR:0, LOCAL:0, LOCAL:1)

TSTR:会話モード '= ""
IF FLAG:調教モード == 調教_居住区
	IF C_COND("行動:買物中",, MTAR)
		TSTR:会話モード '= "商談"
		CALL SKILL_GET_LEVEL(MPLY:0, スキル_大分類_性技, "SEDUCTION")
		SHOPPER_SEDUCTION_LV = RESULT
	ELSEIF C_COND("行動:買春中",, MTAR)
		IF IS_SEXUAL(MPLY, MTAR)
			TSTR:会話モード '= "買春"
			CALL SKILL_GET_LEVEL(MPLY:0, スキル_大分類_性技, "SEDUCTION")
			SHOPPER_SEDUCTION_LV = RESULT
		ENDIF
	ENDIF
ENDIF

;●プレイヤー側の処理
;固定で獲得するソース
SOURCE:(MPLY:0):歓楽 = 30

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MPLY:0, 140)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 50, 50)

;●ターゲット側の処理
;固定で獲得するソース
SOURCE:(MTAR:0):歓楽 = 30

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MTAR:0, 140)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

;相手の機嫌によって効果が変化
LOCAL:2 = MAX(PALAM:(MTAR:0):怒外, PALAM:(MTAR:0):哀外, PALAM:(MTAR:0):怖外)
LOCAL:3 = MAX(PALAM:(MTAR:0):怒主, PALAM:(MTAR:0):哀主, PALAM:(MTAR:0):怖主)
IF LOCAL:2 >= 100
	;	;<孤高> or <抑圧> があり依存度が300未満の場合
	;	IF (TALENT:(MTAR:0):孤高 || TALENT:(MTAR:0):抑圧) && CFLAG:(MTAR:0):依存度 < 300
	;		;必ず失敗
	;		TFLAG:コマンド成否 = -1
	;		;地の文用にフラグを設定
	;		TFLAG:17 = 4
	IF LOCAL:2 > LOCAL:3
		IF LOCAL:2 == PALAM:(MTAR:0):怒外
			;愚痴を聴く
			SOURCE:(MTAR:0):優越 += 50
			TIMES SOURCE:(MTAR:0):歓楽, 2.00
			SOURCE:(MTAR:0):解消 += MIN(LOCAL:2, 400)
			;地の文用にフラグを設定
			TFLAG:17 = 1
		ELSE
			;話を聴く
			SOURCE:(MTAR:0):愛情 += 30
			CALL ADD_SOURCE_AIZYOU(MTAR:0, 50)
			SOURCE:(MTAR:0):解消 += MIN(LOCAL:2, 400)
			;地の文用にフラグを設定
			TFLAG:17 = 2
		ENDIF
		;必ず成功
		IF TFLAG:コマンド成否 == -1
			TFLAG:コマンド成否 = 0
		ENDIF
	ELSE
		;必ず失敗
		TFLAG:コマンド成否 = -1
		;地の文用にフラグを設定
		TFLAG:17 = 3
		SIF (TALENT:(MTAR:0):孤高 || TALENT:(MTAR:0):抑圧)
			TFLAG:17 = 4
	ENDIF
ENDIF

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	SOURCE:(MTAR:0):不満 += 300
	TFLAG:ムード -= 5
	TCVAR:(MTAR:0):HOムード値 -= 5
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	TFLAG:ムード += 2
	TCVAR:(MTAR:0):HOムード値 += 2
	;大成功
ELSE
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TFLAG:ムード += 10
	TCVAR:(MTAR:0):HOムード値 += 10
ENDIF

IF TSTR:会話モード == "商談"
	SELECTCASE TFLAG:コマンド成否
		CASE -1
		CASE 0
			CFLAG:(MTAR:0):キャラ購買意欲 += RAND(3, 7)
		CASEELSE
			CFLAG:(MTAR:0):キャラ購買意欲 += RAND(7, 14)
			TIMES CFLAG:(MTAR:0):キャラ消費金額, 1.25
	ENDSELECT
	IF CFLAG:(MTAR:0):スケベモブ && RAND:100 <= SHOPPER_SEDUCTION_LV * 10
		MAP_SET "HOUSING_MAP", "EVENTCOMEND_CALLEE_ONCE", "HO_TRADE_SEDUCTION_EVENT"
		MAP_SET "HOUSING_MAP", "PLAYER_CHARA_ID", @"%TOSTR(MPLY:0)%"
		MAP_SET "HOUSING_MAP", "TARGET_CHARA_ID", @"%TOSTR(MTAR:0)%"
	ENDIF
ELSEIF TSTR:会話モード == "買春"
	SELECTCASE TFLAG:コマンド成否
		CASE -1
		CASE 0
			SOURCE:(MTAR:0):性行動 = 25
		CASEELSE
			SOURCE:(MTAR:0):性行動 = 40
	ENDSELECT
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

CALL COM_SLG_TRAIN_MAIN_PROCESS(GETNUM(ABL, "話術"), 0)

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE300
#DIM DYNAMIC 机下フェラ
#DIM DYNAMIC COM_PLAYER
#DIM  DYNAMIC CHARA_ID
#DIM  DYNAMIC ROOM_IDX
#DIM  DYNAMIC MODULE_IDX
#DIMS DYNAMIC ROOM_EFFECT_MAP
#DIMS DYNAMIC MODULE_EFFECT_MAP
#DIMS DYNAMIC 設備名
#DIMS DYNAMIC 属性ジャンル

ROOM_IDX = C_COND_MAP_GET(CHARA_ID, "エリアIDX")
ROOM_EFFECT_MAP '= @"MAP_{ROOM_IDX}_ROOM_EFFECT"
MODULE_IDX = C_COND_MAP_GET(CHARA_ID, "設備IDX")
MODULE_EFFECT_MAP '= @"MAP_{MODULE_IDX}_MODULE_EFFECT"
設備名 = %C_COND_MAP_GETS(MPLY, "設備名")%

IF TSTR:会話モード == "商談"
	;机下フェラ判定
	CALL HO_CHECK_USE_MODULE_ELEMENT(MPLY:0, "机")
	IF RESULT
		COM_PLAYER = GET_EQUIP_PLAYER(MPLY:0, _LIST_COM_ORAL)
		SIF COM_PLAYER != -1
			机下フェラ = 1
	ENDIF
	LOCALS '= @"%ANAME(MPLY:0)%は\@机下フェラ ? @C:{カラー_ピンク}@%ANAME(COM_PLAYER)%に%語彙_ペニス()%%S_R("を舐め回させ/をしゃぶらせ/に奉仕させ")%ながら@/C@、平静を装って # \@%ANAME(MTAR:0)%"
	CALLF TAG_PRINT(@"%LOCALS%%S_R("に雑談を交えつつ商品を売り込んだ/と商品の話をした/に世間話を交えつつ商品を売り込んだ")%", 1)
	SELECTCASE TFLAG:コマンド成否
		CASE -1
			PRINTFORML しかし%ANAME(MTAR:0)%は%S_R("あまり興味がないようだ/退屈そうに聞き流している/適当に相槌をうって聞き流している")%…
		CASE 0
			PRINTFORML %ANAME(MTAR:0)%は%S_R(@"商品に顔を近づけてしげしげと眺めている/%ANAME(MPLY:0)%の話を興味深げに聞いている/商品を手にとって眺めている")%
		CASEELSE
			PRINTFORML %ANAME(MTAR:0)%は%S_R(@"商品を食い入るように眺めている/%ANAME(MPLY:0)%の話に興味津々だ/商品を手にとってあれこれ%ANAME(MPLY:0)%に尋ねている")%
	ENDSELECT
	PRINTL
	RETURN
ENDIF
IF TSTR:会話モード == "買春"
	PRINTFORMW %ANAME(MPLY)%は%ANAME(MTAR)%に自身を買わないか持ちかけた
	SELECTCASE TFLAG:コマンド成否
		CASE -1
			PRINTFORML but %ANAME(MTAR)%%S_R(@"doesn't seem to be very interested/I was treated appropriately")%…
		CASE 0
			PRINTFORML %ANAME(MTAR)%は%S_R(@"%ANAME(MPLY)% is looking closely at the face and specifying the product / %ANAME(MPLY)% staring at my body")%
		CASEELSE
			PRINTFORML %ANAME(MTAR)%は%S_R(@"%ANAME(MPLY)%It's nailed to the body/With a nice smile, %ANAME(MPLY)% stars at my body")%
	ENDSELECT
	PRINTL
	RETURN
ENDIF
IF TFLAG:17 == 1
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の愚痴を聴いてやった
		PRINTFORML %ANAME(MTAR:0)%s anger seems to have subsided a little ...
		;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に延々と愚痴を聞かされた
		PRINTFORML %ANAME(MTAR:0)%s anger seems to have subsided a little ...
	ENDIF
ELSEIF TFLAG:17 == 2
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の悩みを聴くことにした
		PRINTFORML %ANAME(MTAR:0)% has calmed down a little...
		;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に悩みを打ち明けた
		PRINTFORML %ANAME(MTAR:0)% has calmed down a little...
	ENDIF
ELSEIF TFLAG:17 == 3
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の話を聞こうとしたが、断られてしまった
	PRINTFORML When the conversation was interrupted, an awkward air flowed ...
ELSEIF TFLAG:17 == 4
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の話を聞こうとしたが、断られてしまった
	PRINTFORML When the conversation was interrupted, an awkward air flowed ...
ELSE
	IF C_COND("設備属性:狭隘",, MPLY)
		属性ジャンル '= "狭隘"
	ELSE
		属性ジャンル '= "着席"
	ENDIF
	{
		CALLF PRINT_RANDC(
			TEXT_CONCATEC(
			@"%ANAME(MPLY)%は%ANAME(MTAR)%と", 1
			,@"%S_R(@"%設備名%に腰掛けて/%設備名%に座って")%", C_COND("設備属性:椅子",, MPLY) && 属性ジャンル == "着席"
			,@"%S_R(@"腰掛けて/座って")%、", C_COND("設備属性:机,設備属性:卓","OR", MPLY)
			,@"%S_R(@"シーツの上で/ベッドで/ベッドの上で/ベッドに腰掛けて")%", C_COND("設備属性:ベッド",, MPLY) && 属性ジャンル == "着席"
			,@"%S_R(@"間近/吐息がかかるほどの距離/肌が触れるほどの距離/窮屈な空間/%設備名%/窮屈な%設備名%")%で、", 属性ジャンル == "狭隘"
			,@"%S_R("指先を絡ませながら/隣りあって/少し照れながら/じゃれあいながら/微笑みを交わしながら/見つめあいながら")%", TALENT:MTAR:恋人
			,@"%S_R("世間話/とりとめのない話/雑談/たわいない話")%をした", 1
			), 1
			)
	}
	PRINTW
ENDIF
IF TFLAG:コマンド成否 == 1
	PRINTFORML 楽しい時間が流れた…
ELSEIF TFLAG:コマンド成否 == -1
	PRINTFORML 会話が途切れると気まずい空気が流れた…
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST300
;日常コマンドに共通の地の文
IF TSTR:会話モード != "商談"
	CALL COM_TEXT_LAST_LIFE
ELSE
	CALL COM_TEXT_LAST_LIFE_GUEST
	IF CFLAG:(MTAR:0):キャラ消費金額 < 1000
		PRINTFORML %ANAME(MTAR:0)%は%S_R("どうやらあまり懐に余裕がないようだ/財布を広げて寂しそうな顔をした/財布を見つめてため息をついた")%
	ELSEIF CFLAG:(MTAR:0):キャラ購買意欲 < 10
		PRINTFORML %ANAME(MTAR:0)%%S_R("はもうあまり物を買う気はないようだ/は冷やかし半分に陳列棚を眺めている/の足は帰り道の方に向いている")%
	ENDIF
	PRINTL
	;予算表示
	LOCALS '= "★" * MAX(1, (CFLAG:(MTAR:0):キャラ消費金額 / 3000))
	CALLF TAG_PRINT(@"@C:{カラー_白}@%ANAME(MTAR:0)%'s budget：@/C@", 0)
	CALLF TAG_PRINT(@"@C:{カラー_黄}@%LOCALS%@/C@" , 1)
	;購買意欲表示
	LOCALS '= "❤" * MAX(1, (CFLAG:(MTAR:0):キャラ購買意欲 / 10))
	CALLF TAG_PRINT(@"@C:{カラー_白}@%ANAME(MTAR:0)%'s motivation to buy:@/C@", 0)
	CALLF TAG_PRINT(@"@C:{カラー_緑}@%LOCALS%@/C@" , 1)
ENDIF

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN300
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF

repo_name: pervysquirrelz/erafl

license: unknown


code:
;警邏

;-------------------------------------------------
;コマンド名称
;------------------------------------------------
@COM_NAME305
RESULTS:0 = 警邏に出掛ける
RESULTS:1 = 警邏に連れ出される

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE305
;共通部分
CALL COM_ABLE_COMMON(305)
SIF RESULT == 0
	RETURN 0
;捕虜会話では不可
SIF FLAG:調教モード == 6
	RETURN 0
;主人公がプレイヤーのとき限定
SIF MPLY:0 != MASTER
	RETURN 0
;デート中は無理
SIF TFLAG:デート場所
	RETURN 0
;臨月、育児、怪我なら不可
SIF GROUPMATCH(CFLAG:(MTAR:0):行動不能状態, 1, 2, 3)
	RETURN 0
;捕虜なら不可
SIF CFLAG:(MTAR:0):捕虜先
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM305
;コマンドの成否をTFLAG:コマンド成否にセット
CALL JUDGE_COM_RESULT(MTAR:0, 5, 5)

;固定で獲得するソース
SOURCE:(MTAR:0):歓楽 = 20

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MTAR:0, 160)

;このターンの経済成長パワー追加値
LOCAL:0 = (ABL:(MPLY:0):身体 + ABL:(MTAR:0):身体 + ABL:(MPLY:0):社交 + ABL:(MTAR:0):社交 + 200) * 25

;このターンの兵数増強パワー追加値
LOCAL:1 = (ABL:(MPLY:0):身体 + ABL:(MTAR:0):身体 + ABL:(MPLY:0):社交 + ABL:(MTAR:0):社交 + 200) * 2 / 5

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 100, 100)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 100, 100)

;ターゲットがサボり魔なら効果減少
IF TALENT:(MTAR:0):サボり魔
	TIMES SOURCE:(MTAR:0):歓楽, 0.80
	SOURCE:(MTAR:0):反感 = 70 - SOURCE:(MTAR:0):歓楽 / 3
ENDIF

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	SOURCE:(MTAR:0):不満 += 300
	TIMES LOCAL:0, 0.20
	TIMES LOCAL:1, 0.20
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	;大成功
ELSE
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES LOCAL:0, 2.00
	TIMES LOCAL:1, 2.00
ENDIF

SELECTCASE FLAG:経済成長追加値
	CASE IS > 100000
		TIMES LOCAL:0, 0.10
	CASE IS > 50000
		TIMES LOCAL:0, 0.25
	CASE IS > 25000
		TIMES LOCAL:0, 0.5
ENDSELECT

SELECTCASE FLAG:兵数増加追加値
	CASE IS > 3000
		TIMES LOCAL:1, 0.10
	CASE IS > 2000
		TIMES LOCAL:1, 0.25
	CASE IS > 1000
		TIMES LOCAL:1, 0.5
ENDSELECT

FLAG:経済成長追加値 += LOCAL:0
FLAG:兵数増加追加値 += LOCAL:1

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE305
;主導権が主人公側
IF IS_INITIATIVE(MASTER)
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%と街の見回りを行った
ELSE
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に引きずられるようにして街の見回りへと出掛けた
ENDIF

IF TFLAG:コマンド成否 == 1
	PRINTFORML なんだかデートしているような雰囲気だ…
ELSEIF TFLAG:コマンド成否 == -1
	PRINTFORML ……いざこざに巻き込まれて散々な目に会った
ENDIF
PRINTFORML (経済と兵数の増加量にボーナス！)
;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST305
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN305
RETURN コマンド_日常


code:
;料理

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME307
RESULTS:0 = 料理に付き合わせる
RESULTS:1 = 料理に付き合わされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE307
;共通部分
CALL COM_ABLE_COMMON(307)
SIF RESULT == 0
	RETURN 0
;捕虜会話では不可
SIF FLAG:調教モード == 6
	RETURN 0
;主人公がプレイヤーのとき限定
SIF MPLY:0 != MASTER
	RETURN 0
;デート中は無理
SIF TFLAG:デート場所
	RETURN 0
;臨月or怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 行動不能_臨月 || CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM307
CALL COM_SLG_TRAIN_MAIN_PROCESS(GETNUM(ABL, "料理"), 0)

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE307
;主導権が主人公側
IF IS_INITIATIVE(MASTER)
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%と一緒に料理を作った
ELSE
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%と一緒に料理を作ることになった
ENDIF

IF ABL:(MPLY:0):料理 < 0 || ABL:(MTAR:0):料理 < 0
	IF TFLAG:コマンド成否 == 1
		PRINTFORML どうにか人間でも食べられる物に仕上がった…
	ELSEIF TFLAG:コマンド成否 == -1
		IF RAND:2 == 0
			PRINTFORML ……臭いだけで気絶しそうなシロモノが完成した
		ELSE
			PRINTFORML ……皿の中に何やら怪しく蠢いているものが見える
		ENDIF
	ELSE
		PRINTFORML 精神修行だと思えば食べられなくもない……と思う
	ENDIF
ELSE
	IF TFLAG:コマンド成否 == 1
		PRINTFORML 思った以上に美味しく仕上がった…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML ……見事な失敗作が出来上がった
	ENDIF
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST307
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN307
RETURN コマンド_日常


code:
;猥談

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME309
RESULTS:0 = エッチな話を振る
RESULTS:1 = エッチな話を振られる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE309
;共通部分
CALL COM_ABLE_COMMON(309)
SIF RESULT == 0
	RETURN 0
;主人公がプレイヤーのとき限定
SIF MPLY:0 != MASTER
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
;プレイヤーとターゲットの性知識が共にLv1以上ないと不可
SIF ABL:(MPLY:0):性知識 < 1 || ABL:(MTAR:0):性知識 < 1
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM309
;実行値の計算
CALL COM_ORDER_309(MTAR:0)
LOCAL:5 = RESULT

;成功
IF LOCAL:5 >= 0
	TFLAG:17 = 1

	SOURCE:(MTAR:0):歓楽 = 20 + MIN(LOCAL:5 / 2, 30)
	;親密に応じた歓楽のソース追加
	CALL ADD_SOURCE_KANRAKU(MTAR:0, 100 + MIN(LOCAL:5 / 3, 60))
	SOURCE:(MTAR:0):感応 = 50 + MIN(LOCAL:5, 100)

	;ムードの増加
	TFLAG:ムード += 4 + MIN(LOCAL:5 / 10, 16)

	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

	LOCAL:2 = 10
	IF TALENT:(MPLY:0):好奇心
		TIMES LOCAL:2, 1.50
	ENDIF
	IF TALENT:(MPLY:0):習得早い
		TIMES LOCAL:2, 1.50
	ENDIF
	IF TALENT:(MPLY:0):習得遅い
		TIMES LOCAL:2, 0.80
	ENDIF

	LOCAL:3 = 10
	IF TALENT:(MTAR:0):好奇心
		TIMES LOCAL:2, 1.50
	ENDIF
	IF TALENT:(MTAR:0):習得早い
		TIMES LOCAL:2, 1.50
	ENDIF
	IF TALENT:(MTAR:0):習得遅い
		TIMES LOCAL:2, 0.80
	ENDIF

	IF ABL:(MPLY:0):性知識 <= 2
		EXP:(MPLY:0):性知識経験値 += LOCAL:2 / ABL:(MPLY:0):性知識
	ENDIF
	IF ABL:(MTAR:0):性知識 <= 2
		EXP:(MTAR:0):性知識経験値 += LOCAL:3 / ABL:(MPLY:0):性知識
	ENDIF

;失敗
ELSE
	TFLAG:17 = 0

	SOURCE:(MTAR:0):反感 = 150 + LIMIT(-LOCAL:5 * 5, 0, 600)
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = 1

RETURN 1

;-------------------------------------------------
;成否判定
;-------------------------------------------------
@COM_ORDER_309(ARG:0)
CVARSET TCVAR, 24, -10000
CVARSET TCVAR, 25, 10000

TCVAR:(ARG:0):実行値 = 80

CALL COM_ORDER_BASE(ARG:0)

IF TALENT:(ARG:0):反抗的
	CALL COM_ORDER_ELEMENT(ARG:0, "反抗的", -5)
ENDIF
IF TALENT:(ARG:0):プライド高い
	CALL COM_ORDER_ELEMENT(ARG:0, "プライド高い", -15)
ENDIF
IF TALENT:(ARG:0):プライド低い
	CALL COM_ORDER_ELEMENT(ARG:0, "プライド低い", 5)
ENDIF
IF TALENT:(ARG:0):自制心
	CALL COM_ORDER_ELEMENT(ARG:0, "自制心", -8)
ENDIF
IF TALENT:(ARG:0):無関心
	CALL COM_ORDER_ELEMENT(ARG:0, "無関心", -6)
ENDIF
IF TALENT:(ARG:0):好奇心
	CALL COM_ORDER_ELEMENT(ARG:0, "好奇心", 15)
ENDIF
IF TALENT:(ARG:0):保守的
	CALL COM_ORDER_ELEMENT(ARG:0, "保守的", -10)
ENDIF
IF TALENT:(ARG:0):抑圧
	CALL COM_ORDER_ELEMENT(ARG:0, "抑圧", -10)
ENDIF
IF TALENT:(ARG:0):孤高
	CALL COM_ORDER_ELEMENT(ARG:0, "孤高", -(20 * KOKOU_RATE(ARG:0) / 100))
ENDIF
IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -10)
ENDIF
IF TALENT:(ARG:0):恥薄い
	CALL COM_ORDER_ELEMENT(ARG:0, "恥薄い", 8)
ENDIF
IF TALENT:(ARG:0):快感に素直
	CALL COM_ORDER_ELEMENT(ARG:0, "快感に素直", 10)
ENDIF
IF TALENT:(ARG:0):快感の否定
	CALL COM_ORDER_ELEMENT(ARG:0, "快感の否定", -35)
ENDIF
IF GETBIT(TALENT:(ARG:0):淫乱系, 素質_淫乱_淫乱)
	CALL COM_ORDER_ELEMENT(ARG:0, "淫乱", 20)
ENDIF

IF IS_SAMESEX(MPLY:0, ARG:0)
	CALL COM_ORDER_ELEMENT(ARG:0, "同性", 50)
ELSE
	IF TALENT:(ARG:0):一線越えない
		CALL COM_ORDER_ELEMENT(ARG:0, "一線越えない", -15)
	ENDIF
ENDIF

IF TALENT:(ARG:0):親愛
	CALL COM_ORDER_ELEMENT(ARG:0, "親愛", 25)
ELSEIF TALENT:(ARG:0):恋慕
	CALL COM_ORDER_ELEMENT(ARG:0, "恋慕", 10)
ENDIF

IF TALENT:(ARG:0):恋人
	CALL COM_ORDER_ELEMENT(ARG:0, "恋人", 20)
ELSEIF TALENT:(ARG:0):親友
	CALL COM_ORDER_ELEMENT(ARG:0, "親友", 20)
ENDIF

IF TALENT:(ARG:0):隷属
	CALL COM_ORDER_ELEMENT(ARG:0, "隷属", 25)
ELSEIF TALENT:(ARG:0):服従
	CALL COM_ORDER_ELEMENT(ARG:0, "服従", 10)
ENDIF

IF TALENT:(ARG:0):烙印
	CALL COM_ORDER_ELEMENT(ARG:0, "烙印", 20)
ENDIF

IF TALENT:(ARG:0):所有者
	CALL COM_ORDER_ELEMENT(ARG:0, "所有者", 25)
ELSEIF TALENT:(ARG:0):主人
	CALL COM_ORDER_ELEMENT(ARG:0, "主人", 10)
ENDIF

CALL COM_ORDER_ELEMENT(ARG:0, "合意", 15)

IF TALENT:(ARG:0):プライド高い
	LOCAL:0 = 4
ELSEIF TALENT:(ARG:0):プライド低い
	LOCAL:0 = 1
ELSE
	LOCAL:0 = 2
ENDIF
IF MARK:(ARG:0):既成事実 >= 1
	CALL COM_ORDER_ELEMENT(ARG:0, @"既成事実Lv{MARK:(ARG:0):既成事実}", MARK:(ARG:0):既成事実 * 3 * LOCAL:0)
ENDIF

IF TALENT:MASTER:魅惑
	CALL COM_ORDER_ELEMENT(ARG:0, "主人公の魅惑", 6)
ENDIF
IF TALENT:MASTER:謎の魅力
	CALL COM_ORDER_ELEMENT(ARG:0, "主人公の謎の魅力", 6)
ENDIF

IF PALAM:(ARG:0):酔い >= 150
	CALL COM_ORDER_ELEMENT(ARG:0, "酔い", PALAM:(ARG:0):酔い / 150)
ENDIF

;判定結果を表示
PRINTFORMW  = {TCVAR:(ARG:0):判定値} %TOSTR_COMPARE(TCVAR:(ARG:0):判定値, TCVAR:(ARG:0):実行値)% 実行値{TCVAR:(ARG:0):実行値}

RETURN TCVAR:(ARG:0):判定値 - TCVAR:(ARG:0):実行値

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE309
;成功
IF TFLAG:17
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%とエッチな話で盛り上がった…
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%にエッチな話題を振ってからかった…
	ENDIF
;失敗
ELSE
	PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%にエッチな話題を振ったが、%ANAME(MTAR:0)%に白い目で見られてしまった…
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST309
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN309
RETURN コマンド_日常


code:
;正常位

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME30

IF IS_INSERT_SINGLE(MTAR:0, "Ａ") && GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 4, 6, 30) && IS_INSERT_MUTUAL(MPLY:0, MTAR:0) != 2
	LOCALS:0 = 二穴正常位
ELSE
	LOCALS:0 = 正常位
ENDIF
RESULTS:0 '= TR(LOCALS:0,"で挿入")
RESULTS:1 '= TR(LOCALS:0,"で挿入させられる")
RESULTS:2 '= TR(LOCALS:0,"で挿入させる")
RESULTS:3 '= TR(LOCALS:0,"で挿入される")
RESULTS:4 '= TR(LOCALS:0,"で挿入させる")
RESULTS:5 '= TR(LOCALS:0,"見せつけ")
IF SEARCH_EQUIP(20, MPLY:0, MTAR:0) != -1 || SEARCH_EQUIP(20, MTAR:0, MPLY:0) != -1
	IF IS_INSERT_SINGLE(MTAR:0, "Ａ") && GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 4, 6, 30)
		LOCALS:0 = 二穴キスハメ
	ELSE
		LOCALS:0 = キスハメ
	ENDIF

	RESULTS:0 = %LOCALS:0%する
	RESULTS:1 = %LOCALS:0%させられる
	RESULTS:2 = %LOCALS:0%させる
	RESULTS:3 = %LOCALS:0%される
	RESULTS:4 = %LOCALS:0%させる
	RESULTS:5 = %LOCALS:0%見せつけ
ENDIF

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE30
CALL COM_ABLE_COMMON(30)
SIF RESULT == 0
	RETURN 0
SIF MPLY_NUM <= 0 || MPLY_NUM > 1
	RETURN 0
SIF MTAR_NUM <= 0 || MTAR_NUM > 1
	RETURN 0
SIF !HAS_VAGINA(MTAR:0, 1)
	RETURN 0
SIF IS_V_HOLD(MTAR:0)
	RETURN 0
;Target is getting assfucked, unless it's sitting backwards or reverse cowgirl
SIF IS_INSERT_SINGLE(MTAR:0, "Ａ") && !GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 4, 6, 30)
	RETURN 0
SIF !CAN_FUCK(MPLY:0, MTAR:0, 30)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM30
;実行判定
CALL COM_ORDER_COMMON
SIF RESULT == 0
	RETURN 0

;既に挿入中なら現在の体位を取得
TFLAG:17 = 0
IF IS_INSERT_MUTUAL(MPLY:0, MTAR:0) == 1
	SELECTCASE GET_SEX_POSITION(MPLY:0, MTAR:0)
		CASE 1
			TFLAG:17 = 1
		CASE 2
			TFLAG:17 = 3
		CASE 3
			TFLAG:17 = 2
		CASE 4
			TFLAG:17 = 3
		CASE 5
			TFLAG:17 = 2
		CASE 6
			TFLAG:17 = 3
		CASE 7
			TFLAG:17 = 3
		CASE 8
			TFLAG:17 = 1
	ENDSELECT
ENDIF
IF TFLAG:17 == 0
	SELECTCASE PREVCOM
		CASE 30
			TFLAG:17 = 1
		CASE 31
			TFLAG:17 = 3
		CASE 32
			TFLAG:17 = 2
		CASE 33
			TFLAG:17 = 3
		CASE 34
			TFLAG:17 = 2
		CASE 35
			TFLAG:17 = 3
		CASE 37
			TFLAG:17 = 3
		CASE 38
			TFLAG:17 = 1
	ENDSELECT
ENDIF

;挿入関係を一時リセット
CALL CLEAR_INSERT_FLAG(MPLY:0, "Ｐ")

;●プレイヤー側の処理
DOWNBASE:(MPLY:0):体力 += 160

EXP:(MPLY:0):性技経験値 += 1
EXP:(MPLY:0):性交経験値 += 2

SOURCE:(MPLY:0):快Ｐ = SENSE_SEX_PLAYER_P(MTAR:0, MPLY:0, 1500)
SOURCE:(MPLY:0):露出 = 100
SOURCE:(MPLY:0):性行動 = 450

;性交系の共通処理
CALL COM_COMMON_SEX_P(MPLY:0, MTAR:0)

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 150, 80)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MPLY:0):奉仕経験取得可能 = 1

;●ターゲット側の処理
DOWNBASE:(MTAR:0):体力 += 120

EXP:(MTAR:0):性技経験値 += 1
EXP:(MTAR:0):性交経験値 += 2

SOURCE:(MTAR:0):快Ｖ = SENSE_SEX_TARGET(MPLY:0, MTAR:0, 1500)
SOURCE:(MTAR:0):露出 = 200
SOURCE:(MTAR:0):性行動 = 450

;性交系の共通処理
CALL COM_COMMON_SEX_V(MTAR:0, MPLY:0)

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 150, 100)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MTAR:0):奉仕経験取得可能 = 1

;体位フラグのセット
TCVAR:(MPLY:0):31 = 1
TCVAR:(MTAR:0):33 = 1

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 4

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP30
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP30(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;ソースを退避
CALL PUTOUT_SOURCE(LOCAL:2)
CALL PUTOUT_SOURCE(LOCAL:3)

DOWNBASE:(LOCAL:2):体力 += 20

EXP:(LOCAL:2):性技経験値 += 1
EXP:(LOCAL:2):性交経験値 += 1

SOURCE:(LOCAL:2):快Ｐ += SENSE_SEX_PLAYER_P(LOCAL:3, LOCAL:2, 750)
SOURCE:(LOCAL:2):露出 += 50
SOURCE:(LOCAL:2):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_P(LOCAL:2, LOCAL:3)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:2):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:2):倒錯度変化基準値 -= 1

DOWNBASE:(LOCAL:3):体力 += 20

EXP:(LOCAL:3):性技経験値 += 1
EXP:(LOCAL:3):性交経験値 += 1

SOURCE:(LOCAL:3):快Ｖ += SENSE_SEX_TARGET(LOCAL:2, LOCAL:3, 750)
SOURCE:(LOCAL:3):露出 += 100
SOURCE:(LOCAL:3):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_V(LOCAL:3, LOCAL:2)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:3):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:3):倒錯度変化基準値 -= 1

;退避したソースを加算
CALL SUM_SOURCE(LOCAL:2)
CALL SUM_SOURCE(LOCAL:3)

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE30(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%の膣に正常位で挿入中

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP30(ARG:0)
SELECTCASE RAND:2
	CASE 0
		PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%の膣が%EQUIP_PLAYER_ANAME(ARG:0)%の%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を咥え込んでいる…
	CASE 1
		LOCAL:0 = MEQUIP_TARGET:(ARG:0):0
		;ターゲットが行動不能
		IF !IS_PLAYABLE(LOCAL:0)
			PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%はぐったりとしている%EQUIP_TARGET_ANAME(ARG:0)%に挿入したまま、さらに激しく腰を動かしている…
		ELSE
			SELECTCASE GET_COM_INITIATIVE()
					;プレイヤーに主導権
				CASE 0
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%に挿入したまま、さらに激しく腰を動かしている…
					;ターゲットに主導権
				CASE 1
					PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%は%EQUIP_PLAYER_ANAME(ARG:0)%の腰に足を絡めながら、さらに激しく腰を動かすように要求し
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%に求められるまま、何度も腰を打ち付けている…
					;第三者に主導権
				CASEELSE
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%EQUIP_TARGET_ANAME(ARG:0)%に正常位で挿入している%EQUIP_PLAYER_ANAME(ARG:0)%に対し、もっと激しく腰を動かすように命令し
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は言われるがまま、激しく何度も%EQUIP_TARGET_ANAME(ARG:0)%の膣奥を突いている…
			ENDSELECT
		ENDIF
ENDSELECT

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP30(ARG:0)
PRINTFORMW %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の膣内から%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を引き抜いた…
;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER30(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 90

;膣性交共通
CALL COM_ORDER_VSEX_P(ARG:0)
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE30
#DIM  DYNAMIC 被覆衣装部位
#DIM  DYNAMIC 被覆衣装ID
#DIMS DYNAMIC 被覆衣装描写

;下着で秘部が覆われているか確認する
CALL IS_COVERED(MTAR:0, 被覆部位_秘部)
被覆衣装部位 = RESULT:0
被覆衣装ID = RESULT:1
IF 被覆衣装部位 == 衣装部位_下着下 || 衣装部位_上下下着
	TRYCCALLFORM CLOTH_DESC_%CLOTH_PART_NAME_ENG:被覆衣装部位%_{被覆衣装ID}
		被覆衣装描写 '= RESULTS
	CATCH
		TRYCCALLFORM CLOTH_NAME_%CLOTH_PART_NAME_ENG:被覆衣装部位%_{被覆衣装ID}
			被覆衣装描写 '= RESULTS
		CATCH
		ENDCATCH
	ENDCATCH
ELSE
	被覆衣装描写 '= ""
ENDIF

IF IS_PIERCED(MTAR:0, ピアス_ラビア)
	LOCALS:0 = ピアスの取り付けられた
ELSE
	LOCALS:0 =
ENDIF

IF TALENT:(MTAR:0):処女 == 1
	LOCALS:0 = まだ乙女の
ELSEIF PALAM:(MTAR:0):潤滑 >= PALAMLV:7
	LOCALS:0 = 濡れそぼった
ELSE
	LOCALS:0 =
ENDIF
IF IS_OVULATION(MTAR:0)
	LOCALS:0 = %LOCALS:0%危険日の
ENDIF

IF TALENT:(MTAR:0):体格 == 体格_小柄
	LOCALS:0 = %LOCALS:0%幼いワレメ
ELSE
	LOCALS:0 = %LOCALS:0%秘裂
ENDIF
IF SEARCH_EQUIP(20, MPLY:0, MTAR:0) != -1 || SEARCH_EQUIP(20, MTAR:0, MPLY:0) != -1
	LOCALS:1 = 口づけを交わしながら
ELSE
	LOCALS:1 =
ENDIF

IF PALAM:(MTAR:0):欲情 >= PALAMLV:10 || (PALAM:(MTAR:0):欲情 >= PALAMLV:7 && TCVAR:(MTAR:0):媚薬)
	IF !TFLAG:17
		SELECTCASE RAND:2
			CASE 0
				PRINTFORML 発情しきった%ANAME(MTAR:0)%の視線は%ANAME(MPLY:0)%の%BAR_NAME(MPLY:0)%に釘付けになっている…
			CASE 1
				PRINTFORML 情欲の虜になった%ANAME(MTAR:0)%は、%ANAME(MPLY:0)%の%BAR_NAME(MPLY:0)%に物欲しげな視線を向けている…
		ENDSELECT
	ELSEIF ABL:(MTAR:0):Ｖ感 >= ランク閾値:ランク_その他:ランク_C && TFLAG:17 == 1
		SELECTCASE RAND:2
			CASE 0
				PRINTFORML %ANAME(MTAR:0)%は肉欲に流されるまま、大きく股を開いて%BAR_NAME(MPLY:0)%を奥まで咥え込んでいる…
			CASE 1
				PRINTFORML 興奮しきった%ANAME(MTAR:0)%は%ANAME(MPLY:0)%の突き込みに合わせ、艶めかしく腰を揺すっている…
		ENDSELECT
	ENDIF
ENDIF

;ペニスバンド装着中
IF IS_EQUIP_PLAYER(MPLY:0, 50)
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORM %ANAME(MPLY:0)%は%ANAME(MTAR:0)%%SPLIT_RAND("に挿入したまま/の腰を掴み/の脚を抱えると", 1)%、%LOCALS:1%
					SELECTCASE IFRAND("0, 1", 1, "2, 3", ABL:(MPLY:0):性交 >= ランク閾値:ランク_その他:ランク_C)
						CASE 0
							PRINTFORMW さらに激しく腰を%SPLIT_RAND("動かし/打ち付け", 1)%%TARGET_STR("た", "てきた")%…
						CASE 1
							PRINTFORMW さらに激しく張り形を突き入れ%TARGET_STR("た", "てきた")%…
						CASE 2
							PRINTFORMW 巧みな腰使いで弱点を執拗に攻め立て%TARGET_STR("た", "てきた")%…
						CASE 3
							PRINTFORMW 貪るように激しく膣を掻き回し、執拗に責め立て%TARGET_STR("た", "てきた")%…
					ENDSELECT
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%を押し倒し、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
				ENDIF
				;ターゲットに主導権
			CASE 1
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の腰に足を絡めながら、さらに激しく腰を動かすように要求した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に求められるまま、%LOCALS:1%何度も腰を打ち付けて%ANAME(MTAR:0)%の膣奥を突いた…
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は%LOCALS:1%自ら背中から倒れ込み、下から%ANAME(MPLY:0)%を迎える格好で正常位で腰を振らせた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の正面に向き直ると、%LOCALS:1%自ら下になって再び%ANAME(MPLY:0)%に腰を振らせた…
				ENDIF
				;第三者に主導権
			CASEELSE
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に正常位で挿入している%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は言われるまま腰の動きを激しくし、%LOCALS:1%何度も%ANAME(MTAR:0)%の膣奥を突いた…
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を正常位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%を押し倒し、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を正常位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒すと、%LOCALS:1%装着した張り型を%ANAME(MTAR:0)%の%LOCALS:0%に突き入れ%TARGET_STR("た", "てきた")%…
				;ターゲットに主導権
			CASE 1
				PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の下になって、%LOCALS:1%自身の%LOCALS:0%へと張り型を導いて挿入させた…
				;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、正常位で%ANAME(MTAR:0)%を犯すように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒すと、%LOCALS:1%装着した張り型を%ANAME(MTAR:0)%の%LOCALS:0%に突き入れた…
		ENDSELECT
	ENDIF

	;ペニスバンド不使用
ELSE
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORM %ANAME(MPLY:0)%は%ANAME(MTAR:0)%%SPLIT_RAND("に挿入したまま/の腰を掴み/の脚を抱えると", 1)%、%LOCALS:1%
					SELECTCASE IFRAND("0, 1", 1, "2, 3", ABL:(MPLY:0):性交 >= ランク閾値:ランク_その他:ランク_C)
						CASE 0
							PRINTFORMW さらに激しく腰を動かし%TARGET_STR("た", "てきた")%…
						CASE 1
							PRINTFORMW さらに激しくペニスを突き入れ%TARGET_STR("た", "てきた")%…
						CASE 2
							PRINTFORMW 巧みな腰使いで弱点を執拗に攻め立て%TARGET_STR("た", "てきた")%…
						CASE 3
							PRINTFORMW 猛烈なピストンで膣を掻き回し%TARGET_STR("た", "てきた")%…
					ENDSELECT
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%を押し倒し、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
				ENDIF
				;ターゲットに主導権
			CASE 1
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の腰に足を絡めながら、さらに激しく腰を動かすように要求した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に求められるまま、%LOCALS:1%何度も腰を打ち付けて%ANAME(MTAR:0)%の膣奥を突いた…
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は自ら後ろに倒れ込んで下になり、%LOCALS:1%%ANAME(MPLY:0)%に正常位で腰を振らせた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の正面に向き直ると、%LOCALS:1%自ら下になって再び%ANAME(MPLY:0)%に腰を振らせた…
				ENDIF
				;第三者に主導権
			CASEELSE
				;前回が正常位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に正常位で挿入している%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は言われるまま腰の動きを激しくし、%LOCALS:1%何度も%ANAME(MTAR:0)%の膣奥を突いた…
					;対面座位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を正常位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%を押し倒し、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
					;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を正常位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%LOCALS:1%上から覆い被さって再び腰を動かし始めた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				IF 被覆衣装描写 == "" || 被覆衣装描写 == "なし"
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒すと、%ANAME(MTAR:0)%の%LOCALS:0%にペニスを突き入れ%LOCALS:1%腰を動かし%TARGET_STR("た", "てきた")%…
				ELSE
					PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒すと%被覆衣装描写%を押しのけ、
					PRINTFORMW 下着の下からあらわになった%LOCALS:0%にペニスを突き入れ、%LOCALS:1%腰を動かし%TARGET_STR("た", "てきた")%…
				ENDIF
				;ターゲットに主導権
			CASE 1
				IF 被覆衣装描写 == "" || 被覆衣装描写 == "なし"
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の下になって、%LOCALS:1%ペニスを自身の%LOCALS:0%へと導いて挿入させた…
				ELSE
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の下になると、%被覆衣装描写%の下で%LOCALS:0%へと%LOCALS:1%ペニスを導いて挿入させた…
				ENDIF
				;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、正常位で%ANAME(MTAR:0)%を犯すように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒すと、%LOCALS:1%ペニスを%ANAME(MTAR:0)%の%LOCALS:0%に突き入れて腰を動かした…
		ENDSELECT
	ENDIF
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST30
SIF IS_PIERCED(MPLY:0, ピアス_ペニス)
	PRINTFORML %ANAME(MPLY:0)%のペニスに取り付けられたピアスが、%ANAME(MTAR:0)%の膣内をごりごりと抉っていく……

IF NOWEX:(MTAR:0):Ｖ絶頂 >= 8
	SELECTCASE RAND:2
		CASE 0
			PRINTFORML %ANAME(MTAR:0)%は汗にまみれた%STR_BODY("身体", MTAR:0)%を震わせながら、蕩けた雌の顔で快楽を受け止めている…
		CASE 1
			PRINTFORML %ANAME(MTAR:0)%は弓なりになってビクビクと痙攣しながら、強烈な絶頂を全身で味わっている…
	ENDSELECT
ENDIF

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN30
RETURN コマンド_ウフフ

;--------------------
;性嗜好用関数
;--------------------
@COM_PREFERENCE_PLAYER_30()
RETURN POWER(2, 性的嗜好_性交したい)

@COM_PREFERENCE_TARGET_30()
RETURN POWER(2, 性的嗜好_性交されたい)

@COM_STACK_SPERM_MPLY_TO_MTAR_30()
RETURN 射精部位_膣内


code:
;酒を飲む

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME310
RESULTS:0 = 酒を飲ませる
RESULTS:1 = 酌をさせられる
RESULTS:2 = 酌をさせる
RESULTS:3 = 酒を飲まされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE310
;共通部分
CALL COM_ABLE_COMMON(310)
SIF RESULT == 0
	RETURN 0
;1:1限定
SIF MPLY_NUM > 1 || MTAR_NUM > 1
	RETURN 0
; 酒類を持っている必要がある
SIF !(HO_ITEM_STOCK:ITEM_蒸留酒)
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
;飲まされる側が妊娠していたら不可
SIF TALENT:(MTAR:0):妊娠
	RETURN 0
;対象が主体的に行動できる
SIF !IS_PLAYABLE(MTAR)
	RETURN 0
;飲酒可能な場所限定
SIF !C_COND("設備用途:飲酒",,MPLY)
	RETURN 0

RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM310
#DIM 酒配列 = ITEM_蒸留酒
#DIM 飲む酒

IF IS_DRUNK(MTAR) > 1 && HO_OCLOCK() < 1800
	IF !(MTAR == MASTER || TALENT:MTAR:サボり魔 || CFLAG:MTAR:捕虜先 == 所属_プレイヤー)
		PRINTFORMW %ANAME(MTAR)%に陽の高いうちから深酒は出来ないと断られてしまった…
		RETURN
	ENDIF
ENDIF

;コマンドの成否をTFLAG:コマンド成否にセット
CALL JUDGE_COM_RESULT(MTAR:0, 15, 5)

;主導権がMASTERにあれば酒を選択できる
IF TFLAG:主導権所有者 == MASTER
	;酒の選択
	PRINTFORML どの飲み物を\@ MPLY:0 == MASTER ? 勧めますか？ # 飲みますか？ \@
	FOR LOCAL, 0, VARSIZE("酒配列")
		IF HO_ITEM_STOCK:(酒配列:LOCAL) > 0
			CALL GET_ITEM_NAME(ITEM_蒸留酒)
			LOCALS = %"[" + RESULTS + "]", 18, LEFT%
			PRINTBUTTON LOCALS, 酒配列:LOCAL
			PRINT
			PRINTFORML {HO_ITEM_STOCK:(酒配列:LOCAL)}個
		ENDIF
	NEXT
	;酒以外の飲み物
	IF HO_ITEM_STOCK:ITEM_ミルク > 1
		CALL GET_ITEM_NAME(ITEM_ミルク)
		LOCALS = %"[" + RESULTS + "]", 18, LEFT%
		PRINTBUTTON LOCALS, ITEM_ミルク
		PRINT
		PRINTFORML {HO_ITEM_STOCK:ITEM_ミルク}個
	ENDIF
	;特殊な飲み物
	IF HAS_SKILL(MTAR, "CUM_ADDICTION", スキル_大分類_性技)
		IF HO_ITEM_STOCK:ITEM_精液 > 10
			LOCALS = %"[ザーメングラス]", 18, LEFT%
			PRINTBUTTON LOCALS, 1000
			PRINT
			PRINTFORML {HO_ITEM_STOCK:ITEM_精液 / 10}個(精液×10を消費)
		ENDIF
	ENDIF
	PRINTL
	PRINTBUTTON "[BACK]", 9999
	$INPUT_LOOP_DRINK
	BINPUT
	CLEARLINE 1
	PRINTL
	IF RESULT == 9999
		RETURN 0
	ELSEIF FINDELEMENT(酒配列, RESULT) != -1
		飲む酒 = RESULT
		;特殊なオプション
	ELSEIF GROUPMATCH(RESULT, ITEM_ミルク)
		飲む酒 = RESULT
	ELSEIF GROUPMATCH(RESULT, 1000)
		飲む酒 = RESULT
	ELSE
		GOTO INPUT_LOOP_DRINK
	ENDIF
ELSE
	WHILE 1
		LOCAL = 酒配列:(RAND:(VARSIZE("酒配列")))
		IF HO_ITEM_STOCK:(LOCAL) > 0
			飲む酒 = LOCAL
			BREAK
		ENDIF
	WEND
ENDIF

SELECTCASE 飲む酒
	CASE ITEM_蒸留酒
		HO_ITEM_STOCK:飲む酒 --
		SOURCE:(MTAR:0):酔い追加 = RAND(150, 200) * 10
		IF TALENT:(MTAR:0):酒豪
			LOCAL:0 = 220
		ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
			LOCAL:0 = 120
			TFLAG:コマンド成否 = -1
		ELSE
			LOCAL:0 = 190
		ENDIF
		;口の中を綺麗に
		CUM_CUR_AMOUNT:MTAR:射精部位_口 = 0
		FLAG:コマンド汎用 = ITEM_蒸留酒
	CASE ITEM_ミルク
		HO_ITEM_STOCK:飲む酒 --
		LOCAL:0 = 60
		;酔い覚め
		PALAM:(MTAR:0):酔い = MAX(0, PALAM:(MTAR:0):酔い - RAND(150, 200) * 10)
		;口の中を綺麗に
		CUM_CUR_AMOUNT:MTAR:射精部位_口 = 0
		FLAG:コマンド汎用 = ITEM_ミルク
		;ザーメングラス
	CASE 1000
		HO_ITEM_STOCK:ITEM_精液 -= 10
		EXP:MTAR:精愛経験値 += 4
		SOURCE:MPLY:逸脱 = 400
		SOURCE:MTAR:性行動 = 300
		SOURCE:MTAR:逸脱 = 800
		CUM_CUR_AMOUNT:MTAR:射精部位_口 += 100
		TCVAR:MTAR:媚薬 += 1
		FLAG:コマンド汎用 = 1000
		LOCAL:0 = 200

		; CASE GETNUM(ITEM, "粗製ワイン")
		; 	ITEM:粗製ワイン --
		; 	SOURCE:(MTAR:0):酔い追加 = 800
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 280
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 120
		; 		TFLAG:コマンド成否 = -1
		; 	ELSE
		; 		LOCAL:0 = 240
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 2
		; CASE GETNUM(ITEM, "アストロ・フィズ")
		; 	ITEM:アストロ・フィズ --
		; 	SOURCE:(MTAR:0):酔い追加 = 600
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 320
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 260
		; 	ELSE
		; 		LOCAL:0 = 320
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 3
		; CASE GETNUM(ITEM, "ピルスナー")
		; 	ITEM:ピルスナー --
		; 	SOURCE:(MTAR:0):酔い追加 = 800
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 320
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 120
		; 		TFLAG:コマンド成否 = -1
		; 	ELSE
		; 		LOCAL:0 = 270
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 4
		; CASE GETNUM(ITEM, "ライスワイン")
		; 	ITEM:ライスワイン --
		; 	SOURCE:(MTAR:0):酔い追加 = 1800
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 360
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 100
		; 		TFLAG:コマンド成否 = -1
		; 	ELSE
		; 		LOCAL:0 = 300
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 5
		; CASE GETNUM(ITEM, "ブランデー")
		; 	ITEM:ブランデー --
		; 	SOURCE:(MTAR:0):酔い追加 = 1200
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 360
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 260
		; 	ELSE
		; 		LOCAL:0 = 400
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 6
		; CASE GETNUM(ITEM, "ラム酒")
		; 	ITEM:ラム酒 --
		; 	SOURCE:(MTAR:0):酔い追加 = 3000
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 500
		; 		TFLAG:コマンド成否 = 1
		; 	ELSEIF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		; 		LOCAL:0 = 100
		; 		TFLAG:コマンド成否 = -1
		; 	ELSE
		; 		LOCAL:0 = 360
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 7
		; CASE GETNUM(ITEM, "ウォッカ")
		; 	ITEM:ウォッカ --
		; 	SOURCE:(MTAR:0):酔い追加 = 5000
		; 	IF TALENT:(MTAR:0):酒豪
		; 		LOCAL:0 = 500
		; 		TFLAG:コマンド成否 = 1
		; 	ELSE
		; 		LOCAL:0 = 50
		; 		TFLAG:コマンド成否 = -1
		; 	ENDIF
		; 	FLAG:コマンド汎用 = 8
	CASEELSE
		PRINTFORMW 存在しない酒が指定されたよ
		PRINTFORMW 飲む酒:{飲む酒}
		PRINTFORMW 上のメッセージと一緒にスレで報告してね
		RETURN 0
ENDSELECT

;●ターゲット側の処理
;固定で獲得するソース
SOURCE:(MTAR:0):歓楽 = LOCAL:0 / 3

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MTAR:0, LOCAL:0)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 50, 50)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MTAR:0):歓楽, 0.50
	SOURCE:(MTAR:0):不満 += 200
	TFLAG:ムード -= 5
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	TFLAG:ムード += 2
	;大成功
ELSE
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TFLAG:ムード += 10
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

IF !GROUPMATCH(飲む酒, ITEM_乳, 1000)
	SELECTCASE ABL:(MTAR:0):肝臓
		CASE 0
			TIMES SOURCE:(MTAR:0):酔い追加, 1.5
		CASE 1
			TIMES SOURCE:(MTAR:0):酔い追加, 1.30
		CASE 2
			TIMES SOURCE:(MTAR:0):酔い追加, 1.15
		CASE 3
			TIMES SOURCE:(MTAR:0):酔い追加, 1.00
		CASE 4
			TIMES SOURCE:(MTAR:0):酔い追加, 0.85
		CASEELSE
			TIMES SOURCE:(MTAR:0):酔い追加, 0.70
	ENDSELECT
	EXP:(MTAR:0):肝臓経験値 += SOURCE:(MTAR:0):酔い追加 / 100
ELSE
ENDIF

CALL FL_PLAYSE("Put_0.mp3")

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE310
#DIMS DYNAMIC 実行名
#DIMS DYNAMIC 対象名

実行名 = %ANAME(MPLY)%
対象名 = %ANAME(MTAR)%

IF !GROUPMATCH(FLAG:コマンド汎用, ITEM_乳, 1000)
	IF MPLY:0 == MASTER
		PRINTFORMW %実行名%は%対象名%に酌を\@ TFLAG:主導権所有者 == MPLY:0 ? した # させられた \@
		IF TFLAG:コマンド成否 == 1
			IF GROUPMATCH(FLAG:コマンド汎用, 3, 4, 5, 6, 7, 8)
				PRINTFORML %対象名%は珍しい酒を飲めて上機嫌だ…
			ELSE
				PRINTFORML いい感じに酔いが回り、いつも以上に話が盛り上がった…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == -1
			IF GROUPMATCH(FLAG:コマンド汎用, 7, 8)
				PRINTFORML ……%対象名%はあまりの度数の強さにむせてしまい、激しく咳き込んだ
			ELSE
				PRINTFORML %対象名%の口には合わなかったようだ…
			ENDIF
		ENDIF
	ELSE
		PRINTFORMW %対象名%は%実行名%に酌を\@ TFLAG:主導権所有者 == MPLY:0 ? させた # された \@…
	ENDIF
ELSEIF GROUPMATCH(FLAG:コマンド汎用, ITEM_乳)
	CALL GET_ITEM_NAME(FLAG:コマンド汎用)
	PRINTFORML %対象名%はグラスに注がれた%RESULTS%で喉を潤した
	IF PALAM:MTAR:酔い > 0
		PRINTFORMW 少し酔いが覚めたようだ
	ELSE
		WAIT
	ENDIF
ELSEIF 1000
	IF MPLY == MASTER
		PRINTFORML %実行名%は容器を取り出すと%対象名%のグラスにどろりとした精液を注いだ
		PRINTFORMW すっかり精飲の虜になった%対象名%は、おびただしい量の白濁がグラスを満たしていく様を陶然とした様子で見つめている
		PRINTFORML
		PRINTFORMW そしてグラスに口をつけると、%実行名%に見せつけるように、喉を鳴らして飲み干した…
	ELSE
		PRINTFORML %実行名%は容器を取り出すと%対象名%のグラスにどろりとした精液を注いだ
		PRINTFORMW %対象名%はグラスを満たすおびただしい量のザーメンから目を離すことができない
		PRINTFORML
		PRINTFORMW %対象名%はゆっくりとグラスに口をつけ、%実行名%の目の前で飲み干した…
	ENDIF
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST310

IF  !GROUPMATCH(FLAG:コマンド汎用, 1000)
	;日常コマンドに共通の地の文
	CALL COM_TEXT_LAST_LIFE
	;酔いに関する情報を表示
	PRINTL
	PRINTL
	LOCALS '= @"%COLOR_BAR_H(PALAM:MTAR:酔い, 10000, 36, カラー_ほろ酔い, カラー_泥酔, カラー_ダークグレー, 0, @"🍺")%"
	PRINTFORM %ANAME(MTAR)%の酔い加減：
	SELECTCASE IS_DRUNK(MTAR)
		CASE 0
			CALLF TAG_PRINTH(@"@C:{カラー_白}@素面@/C@")
		CASE 1
			CALLF TAG_PRINTH(@"@C:{カラー_ほろ酔い}@ほろ酔い@/C@")
		CASE 2
			CALLF TAG_PRINTH(@"@C:{カラー_酔い}@酔い@/C@")
		CASE 3
			CALLF TAG_PRINTH(@"@C:{カラー_酩酊}@!酩酊!@/C@")
		CASE 4
			CALLF TAG_PRINTH(@"@C:{カラー_泥酔}@!!泥酔!!@/C@")
	ENDSELECT

	HTML_PRINT @"%LOCALS%"
ELSEIF FLAG:コマンド汎用 == 1000
	CALLF TAG_PRINTH(@"@C:{カラー_ピンク}@口内を満たす%語彙_精液()%の感触に%ANAME(MTAR)%は恍惚の笑みを浮かべている。すっかり発情しているようだ…@/C@", 1)
ENDIF
PRINTL

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN310
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;蜂蜜水→エインズの露

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME311
RESULTS:0 = エインズの露を渡す
RESULTS:1 = エインズの露を渡す

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE311
;共通部分
CALL COM_ABLE_COMMON(311)
SIF RESULT == 0
	RETURN 0
;捕虜会話では不可
SIF FLAG:調教モード == 6
	RETURN 0
;主人公がプレイヤーのとき限定
SIF MPLY:0 != MASTER
	RETURN 0
;蜂蜜水を持っている必要がある
SIF !ITEM:エインズの露
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM311
;蜂蜜水を一つ減らす
ITEM:エインズの露 --

;コマンドの成否をTFLAG:コマンド成否にセット
CALL JUDGE_COM_RESULT(MTAR:0, 15, 5)

;●ターゲット側の処理
LOCAL:0 = 200

IF TALENT:(MTAR:0):酒豪
	TIMES LOCAL:0, 0.75
ENDIF

IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
	TIMES LOCAL:0, 1.50
	;絶対に失敗しない
	TFLAG:コマンド成否 = MAX(TFLAG:コマンド成否, 0)
ENDIF

;固定で獲得するソース
SOURCE:(MTAR:0):歓楽 = LOCAL:0 / 3

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MTAR:0, LOCAL:0)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 50, 50)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

;絶頂・尿意限界によるおもらしの設定が有効の場合
IF CONFIG:90 == 1
	IF TCVAR:(MTAR:0):尿残量 < 2900
		;尿残量を増やす
		TCVAR:(MTAR:0):尿残量 += MIN(400, 2900 - TCVAR:(MTAR:0):尿残量)
	ENDIF
	;蜂蜜水による尿意の加算基本値を追加
	TCVAR:(MTAR:0):エインズの露 += 2
ENDIF

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MTAR:0):歓楽, 0.50
	SOURCE:(MTAR:0):不満 += 200
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0
	TFLAG:ムード += 2
;大成功
ELSE
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TFLAG:ムード += 10
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE311
PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%にエインズの露をプレゼントした
IF TFLAG:コマンド成否 == 1
	PRINTFORML 思いのほか喜んでもらえたようだ…
ELSEIF TFLAG:コマンド成否 == -1
	PRINTFORML あまり甘いものを飲む気分ではなかったようだ…
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST311
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN311
RETURN コマンド_日常


code:
;後背位

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME31
IF IS_INSERT_SINGLE(MTAR:0, "Ａ") && GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 30) && IS_INSERT_MUTUAL(MPLY:0, MTAR:0) != 2
	LOCALS:0 = 二穴後背位
ELSE
	LOCALS:0 = 後背位
ENDIF

RESULTS:0 '= TR(LOCALS:0,"で挿入")
RESULTS:1 '= TR(LOCALS:0,"で挿入させられる")
RESULTS:2 '= TR(LOCALS:0,"で挿入させる")
RESULTS:3 '= TR(LOCALS:0,"で挿入される")
RESULTS:4 '= TR(LOCALS:0,"で挿入させる")
RESULTS:5 '= TR(LOCALS:0,"見せつけ")

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE31
CALL COM_ABLE_COMMON(31)
SIF RESULT == 0
	RETURN 0
SIF MPLY_NUM <= 0 || MPLY_NUM > 1
	RETURN 0
SIF MTAR_NUM <= 0 || MTAR_NUM > 1
	RETURN 0
SIF !HAS_VAGINA(MTAR:0, 1)
	RETURN 0
SIF IS_V_HOLD(MTAR:0)
	RETURN 0
SIF !CAN_FUCK(MPLY:0, MTAR:0, 31)
	RETURN 0
;being assfucked
SIF IS_INSERT_SINGLE(MTAR:0, "Ａ") && !GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 30)
	RETURN 0
SIF REACHING_BODY(MTAR, MPLY)
	RETURN 0
SIF REACHING_GROIN(MTAR, MPLY)
	RETURN 0
SIF LICKING_BODY(MTAR, MPLY) || LICKING_BODY(MPLY, MTAR)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM31
;実行判定
CALL COM_ORDER_COMMON
SIF RESULT == 0
	RETURN 0

;既に挿入中なら現在の体位を取得
TFLAG:17 = 0
IF IS_INSERT_MUTUAL(MPLY:0, MTAR:0) == 1
	SELECTCASE GET_SEX_POSITION(MPLY:0, MTAR:0)
		CASE 1
			TFLAG:17 = 3
		CASE 2
			TFLAG:17 = 1
		CASE 3
			TFLAG:17 = 3
		CASE 4
			TFLAG:17 = 2
		CASE 5
			TFLAG:17 = 3
		CASE 6
			TFLAG:17 = 2
		CASE 7
			TFLAG:17 = 2
		CASE 8
			TFLAG:17 = 3
	ENDSELECT
ENDIF
IF TFLAG:17 == 0
	SELECTCASE PREVCOM
		CASE 30
			TFLAG:17 = 3
		CASE 31
			TFLAG:17 = 1
		CASE 32
			TFLAG:17 = 3
		CASE 33
			TFLAG:17 = 2
		CASE 34
			TFLAG:17 = 3
		CASE 35
			TFLAG:17 = 2
		CASE 37
			TFLAG:17 = 2
		CASE 38
			TFLAG:17 = 3
	ENDSELECT
ENDIF

;挿入関係を一時リセット
CALL CLEAR_INSERT_FLAG(MPLY:0, "Ｐ")

;●プレイヤー側の処理
DOWNBASE:(MPLY:0):体力 += 160

EXP:(MPLY:0):性技経験値 += 1
EXP:(MPLY:0):性交経験値 += 2

SOURCE:(MPLY:0):快Ｐ = SENSE_SEX_PLAYER_P(MTAR:0, MPLY:0, 1500)
SOURCE:(MPLY:0):性行動 = 450

IF MTAR:0 == MASTER
	CALL ADD_SOURCE_AIZYOU(MPLY:0, 20)
ELSEIF MPLY:0 != MASTER
	SOURCE:(MPLY:0):露出 += 200
ENDIF

;性交系の共通処理
CALL COM_COMMON_SEX_P(MPLY:0, MTAR:0)

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 200, 80)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MPLY:0):奉仕経験取得可能 = 1

;●ターゲット側の処理
DOWNBASE:(MTAR:0):体力 += 120

EXP:(MTAR:0):性技経験値 += 1
EXP:(MTAR:0):性交経験値 += 2

SOURCE:(MTAR:0):快Ｖ = SENSE_SEX_TARGET(MPLY:0, MTAR:0, 1500)
SOURCE:(MTAR:0):露出 = 400
SOURCE:(MTAR:0):性行動 = 450

IF MPLY:0 == MASTER
	CALL ADD_SOURCE_AIZYOU(MTAR:0, 20)
ELSEIF MTAR:0 != MASTER
	SOURCE:(MTAR:0):露出 += 200
ENDIF

;性交系の共通処理
CALL COM_COMMON_SEX_V(MTAR:0, MPLY:0)

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 120, 180)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MTAR:0):奉仕経験取得可能 = 1

;体位フラグのセット
TCVAR:(MPLY:0):31 = 2
TCVAR:(MTAR:0):33 = 2

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = 1

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 3

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP31
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP31(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;ソースを退避
CALL PUTOUT_SOURCE(LOCAL:2)
CALL PUTOUT_SOURCE(LOCAL:3)

DOWNBASE:(LOCAL:2):体力 += 20

EXP:(LOCAL:2):性技経験値 += 1
EXP:(LOCAL:2):性交経験値 += 1

SOURCE:(LOCAL:2):快Ｐ += SENSE_SEX_PLAYER_P(LOCAL:3, LOCAL:2, 750)
SOURCE:(LOCAL:2):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_P(LOCAL:2, LOCAL:3)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:2):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:2):倒錯度変化基準値 += 1

DOWNBASE:(LOCAL:3):体力 += 20

EXP:(LOCAL:3):性技経験値 += 1
EXP:(LOCAL:3):性交経験値 += 1

SOURCE:(LOCAL:3):快Ｖ += SENSE_SEX_TARGET(LOCAL:2, LOCAL:3, 750)
SOURCE:(LOCAL:3):露出 += 200
SOURCE:(LOCAL:3):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_V(LOCAL:3, LOCAL:2)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:3):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:3):倒錯度変化基準値 += 1

;退避したソースを加算
CALL SUM_SOURCE(LOCAL:2)
CALL SUM_SOURCE(LOCAL:3)

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE31(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%の膣に後背位で挿入中

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP31(ARG:0)
IF TALENT:(MEQUIP_TARGET:(ARG:0):0):処女 == 1
	LOCALS:0 = まだ乙女の
ELSEIF PALAM:(MEQUIP_TARGET:(ARG:0):0):潤滑 >= PALAMLV:7
	LOCALS:0 = 濡れそぼった
ELSE
	LOCALS:0 =
ENDIF
IF IS_OVULATION(MEQUIP_TARGET:(ARG:0):0)
	LOCALS:0 = %LOCALS:0%危険日の
ENDIF

IF TALENT:(MEQUIP_TARGET:(ARG:0):0):体格 == 体格_小柄
	LOCALS:0 = %LOCALS:0%幼いワレメ
ELSE
	LOCALS:0 = %LOCALS:0%秘裂
ENDIF
SELECTCASE RAND:2
	CASE 0
		PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%の膣が%EQUIP_PLAYER_ANAME(ARG:0)%の%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を咥え込んでいる…
	CASE 1
		LOCAL:0 = MEQUIP_TARGET:(ARG:0):0
		;ターゲットが行動不能
		IF !IS_PLAYABLE(LOCAL:0)
			SELECTCASE RAND:2
				CASE 0
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は気絶した%EQUIP_TARGET_ANAME(ARG:0)%の膣を後ろから激しく突いている…
				CASE 1
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%はぐったりしている%EQUIP_TARGET_ANAME(ARG:0)%の%LOCALS:0%に後ろから何度も腰を打ち付けている…
			ENDSELECT
		ELSE
			SELECTCASE GET_COM_INITIATIVE()
					;プレイヤーに主導権
				CASE 0
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は四つん這いの%EQUIP_TARGET_ANAME(ARG:0)%の腰を掴み、激しく膣奥を突いている…
						CASE 1
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は後ろから%EQUIP_TARGET_ANAME(ARG:0)%の%LOCALS:0%に突き入れ、何度も何度も繰り返し腰を打ち付けている…
					ENDSELECT
					;ターゲットに主導権
				CASE 1
					PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%は%EQUIP_PLAYER_ANAME(ARG:0)%に見せつけるように尻を振りながら、もっと激しく腰を動かすように要求し
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の腰を掴みなおすと、後ろから何度も腰を打ち付けて%EQUIP_TARGET_ANAME(ARG:0)%の膣奥を突いている…
						CASE 1
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の腰を掴みなおすと、後ろから何度も腰を打ち付け%LOCALS:0%に突き入れている…
					ENDSELECT
					;第三者に主導権
				CASEELSE
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%EQUIP_TARGET_ANAME(ARG:0)%に後背位で挿入している%EQUIP_PLAYER_ANAME(ARG:0)%に対し、もっと激しく腰を動かすように命令し
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は言われるまま腰の動きを激しくし、後ろから何度も%EQUIP_TARGET_ANAME(ARG:0)%の膣奥を突いている…
						CASE 1
							PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は言われるまま腰の動きを激しくし、後ろから何度も%EQUIP_TARGET_ANAME(ARG:0)%の%LOCALS:0%に突き入れている…
					ENDSELECT
			ENDSELECT
		ENDIF
ENDSELECT

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP31(ARG:0)
#DIM DYNAMIC 対象

対象 = MEQUIP_TARGET:(ARG:0):0
IF WOMB_SPERM_AMOUNT:対象:0 < 50
	PRINTL
	PRINTFORMW %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の膣内から%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を引き抜いた…
ELSE
	PRINTL
	PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の膣内から%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を引き抜いた…
	{
		CALLF PRINT_CONCATE(
			@"%EQUIP_TARGET_ANAME(ARG:0)%"
			,@"%S_R(@"の子宮からこぼれた%語彙_精液()%が糸を引いて垂れた…", WOMB_SPERM_AMOUNT:対象:0 < 100)%"
			,@"%S_R(@"に注ぎ込まれた大量の%語彙_精液()%が子宮から止めどなく溢れた…", WOMB_SPERM_AMOUNT:対象:0 < 300)%"
			,@"%S_R(@"に注ぎ込まれたおびただしい量の%語彙_精液()%が子宮から激しく溢れ出た…", WOMB_SPERM_AMOUNT:対象:0 >= 300)%"
			)
	}
	PRINTW
ENDIF

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER31(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 100

;膣性交共通
CALL COM_ORDER_VSEX_P(ARG:0)
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE31
#DIMS DYNAMIC 実行名
#DIMS DYNAMIC 対象名
#DIMS DYNAMIC 下着描写
#DIMS DYNAMIC 潤滑描写
#DIMS DYNAMIC 性器描写
#DIMS DYNAMIC 設備

下着描写 = %C_COND_MAP_GETS(MTAR:0, "被覆衣装描写_秘部_下着")%
IF IS_ESTRUS(MTAR:0, 8) && IS_WET(MTAR:0)
	潤滑描写 = %S_R("びしょびしょに濡れた/濡れそぼった/すっかりぬめりを帯びた/ぐっしょりとした/とろとろに湿った")%
ENDIF

IF HAS_PENIS(MTAR:0, 1)
	性器描写 = %BAR_NAME(MTAR:0)%
ELSE
	性器描写 = %S_R(@"%STR_BODY("クリ：感度：装飾", MTAR:0)%/%STR_BODY("陰唇：感度：装飾", MTAR:0)%")%
ENDIF

SIF C_COND("設備属性:ベッド",, MPLY:0)
	設備 '= "ベッド"

IF IS_PIERCED(MTAR:0, ピアス_ラビア)
	LOCALS:0 = ピアスの取り付けられた
ELSE
	LOCALS:0 =
ENDIF
IF TALENT:(MTAR:0):処女 == 1
	LOCALS:0 = まだ乙女の
ELSEIF PALAM:(MTAR:0):潤滑 >= PALAMLV:7
	LOCALS:0 = 濡れそぼった
ELSE
	LOCALS:0 =
ENDIF
IF IS_OVULATION(MTAR:0)
	LOCALS:0 = %LOCALS:0%危険日の
ENDIF

IF TALENT:(MTAR:0):体格 == 体格_小柄
	LOCALS:0 = %LOCALS:0%幼いワレメ
ELSE
	LOCALS:0 = %LOCALS:0%秘裂
ENDIF

IF TFLAG:17 == 1 && (PALAM:(MTAR:0):欲情 >= PALAMLV:10 || (PALAM:(MTAR:0):欲情 >= PALAMLV:7 && TCVAR:(MTAR:0):媚薬)) && ABL:(MTAR:0):Ｖ感 >= ランク閾値:ランク_その他:ランク_C
	SELECTCASE RAND:2
		CASE 0
			{
				CALLF PRINT_CONCATE(
					@"%ANAME(MTAR:0)%は"
					,@"%S_R("情欲に流されるまま/快楽に身を任せて/肉欲の虜になって/愛欲に身を委ねて", 1)%、"
					,@"%S_R("ねだるように/期待に満ちた視線とともに/さらなる快楽を求めて", 1)%"
					,@"%S_R("シーツの上で/ベッドの上で", 設備 == "ベッド")%"
					,@"%STR_BODY("尻：外見：修飾", MTAR:0)%尻を突き出している…"
					)
			}
			PRINTL
		CASE 1
			{
				CALLF PRINT_CONCATE(
					@"%ANAME(MTAR:0)%は"
					,@"%S_R("獣の姿勢で/四つん這いで/組み伏せられて/尻を突き出して", 1, "シーツに組み伏せられて/ベッドで尻を突き出して/シーツに顔を沈めて", 設備 == "ベッド")%"
					,@"%BAR_NAME(MPLY:0)%%S_R("に背後から責め立てられ/を繰り返し挿入されながら/を突き込まれながら/の注挿を受けながら/に蹂躙されながら")%、"
					,@"%S_R("発情しきった雌そのものの顔で/口からよだれを垂れ流して/唾液を滴らせて/下肢をくねらせて")%"
					,@"%S_R("喘いでいる/嬌声を上げている/淫らな声を上げている")%…"
					)
			}
			PRINTL
	ENDSELECT
ENDIF

;ペニスバンド装着中
IF IS_EQUIP_PLAYER(MPLY:0, 50)
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				;前回が後背位
				IF TFLAG:17 == 1
					SELECTCASE IFRAND("0, 2", 1, "3", ABL:(MPLY:0):主導度Ｕ >= 500)
						CASE 0
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いにした/四つん這いになった", 1)%%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：長", MTAR:0)%を%SPLIT_RAND("掴むと/鷲掴みにすると/揉みしだきながら/掴み直すと", 1, "抱え込むと", GET_HIPSIZE(MTAR:0) >= 1)%、
						CASE 1
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いにした/四つん這いになった/挿入したまま", 1)%%ANAME(MTAR:0)%の%STR_BODY("腰", MTAR:0)%を%SPLIT_RAND("掴むと/掴み直すと/引き寄せると", 1)%、
						CASE 2
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いで繋がったままの/後背位で繋がったままの/尻を突き出した姿勢の", 1)%%ANAME(MTAR:0)%に背中から覆い被さると、
						CASE 3
							PRINTFORM %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：長", MTAR:0)%を%SPLIT_RAND("割り開いて/押し広げて/掴んで", 1)%%SPLIT_RAND("尻穴を/アナルを/結合部を/繋がった部分を", 1)%%SPLIT_RAND("剥き出しにしながら/眺めながら/観察しながら/丸見えにしながら", 1)%、
					ENDSELECT
					SELECTCASE IFRAND("0, 1", 1, "2", ABL:(MPLY:0):性交 >= ランク閾値:ランク_その他:ランク_C)
						CASE 0
							PRINTFORMW さらに激しく腰を動かして膣奥を突い%TARGET_STR("た", "てきた")%…
						CASE 1
							PRINTFORMW 音を立てて激しく腰を打ち付け%TARGET_STR("た", "てきた")%…
						CASE 2
							PRINTFORMW 張り型で弱い部分を激しく擦り上げ%TARGET_STR("た", "てきた")%…
					ENDSELECT
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒して後ろから覆い被さると、後背位の体勢で再び腰を動かし始めた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を返して四つん這いにすると、後背位の体勢で再び腰を動かし始めた…
				ENDIF
				;ターゲットに主導権
			CASE 1
				;前回が後背位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に見せつけるように%STR_BODY("尻：外見", MTAR:0)%を振りながら、もっと激しく腰を動かすように要求した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：腰：長", MTAR:0)%を掴みなおすと、後ろから何度も腰を打ち付けて%ANAME(MTAR:0)%の膣奥を突いた…
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は自ら前に倒れて四つん這いになると、%ANAME(MPLY:0)%に後背位で腰を振らせた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は体をひねって自ら四つん這いになると、今度は%ANAME(MPLY:0)%に後背位で腰を振らせた…
				ENDIF
				;第三者に主導権
			CASEELSE
				;前回が後背位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に後背位で挿入している%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は言われるまま腰の動きを激しくし、後ろから何度も%ANAME(MTAR:0)%の膣奥を突いた…
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を後背位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒して後ろから覆い被さると、再び腰を動かし始めた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を後背位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を返して四つん這いにすると、再び腰を動かし始めた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%は張り型を後ろから%ANAME(MTAR:0)%の%LOCALS:0%に突き入れ、何度も腰を打ち付け%TARGET_STR("た", "てきた")%…
				;ターゲットに主導権
			CASE 1
				PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に%STR_BODY("尻：外見", MTAR:0)%を向け、後ろから%LOCALS:0%に張り型を挿入させた…
				;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、後背位で%ANAME(MTAR:0)%を犯すように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を四つん這いにすると、装着した張り型を%ANAME(MTAR:0)%の%LOCALS:0%に突き入れて腰を振った…
		ENDSELECT
	ENDIF

	;ペニスバンド不使用
ELSE
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				;前回が後背位
				IF TFLAG:17 == 1
					SELECTCASE IFRAND("0, 2", 1, "3", ABL:(MPLY:0):主導度Ｕ >= 500)
						CASE 0
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いにした/四つん這いになった", 1)%%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：長", MTAR:0)%を%SPLIT_RAND("掴むと/鷲掴みにすると/揉みしだきながら/掴み直すと", 1, "抱え込むと", GET_HIPSIZE(MTAR:0) >= 1)%、
						CASE 1
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いにした/四つん這いになった/挿入したまま", 1)%%ANAME(MTAR:0)%の%STR_BODY("腰", MTAR:0)%を%SPLIT_RAND("掴むと/掴み直すと/引き寄せると", 1)%、
						CASE 2
							PRINTFORM %ANAME(MPLY:0)%は%SPLIT_RAND("四つん這いで繋がったままの/後背位で繋がったままの/尻を突き出した姿勢の", 1)%%ANAME(MTAR:0)%に背中から覆い被さると、
						CASE 3
							PRINTFORM %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：長", MTAR:0)%を%SPLIT_RAND("割り開いて/押し広げて/掴んで", 1)%%SPLIT_RAND("尻穴を/アナルを/結合部を/繋がった部分を", 1)%%SPLIT_RAND("剥き出しにしながら/眺めながら/観察しながら/丸見えにしながら", 1)%、
					ENDSELECT
					SELECTCASE IFRAND("0, 1", 1, "2", ABL:(MPLY:0):性交 >= ランク閾値:ランク_その他:ランク_C)
						CASE 0
							PRINTFORMW さらに激しく腰を動かして膣奥を突い%TARGET_STR("た", "てきた")%…
						CASE 1
							PRINTFORMW 音を立てて激しく腰を打ち付け%TARGET_STR("た", "てきた")%…
						CASE 2
							PRINTFORMW ペニスで弱い部分を激しく擦り上げ%TARGET_STR("た", "てきた")%…
					ENDSELECT
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒して後ろから覆い被さると、後背位の体勢で再び腰を動かし始めた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を返して四つん這いにすると、後背位の体勢で再び腰を動かし始めた…
				ENDIF
				;ターゲットに主導権
			CASE 1
				;前回が後背位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に見せつけるように%STR_BODY("尻：外見", MTAR:0)%を振りながら、もっと激しく腰を動かすように要求した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY("尻：尻肉：腰：長", MTAR:0)%を掴みなおすと、後ろから何度も腰を打ち付けて%ANAME(MTAR:0)%の膣奥を突いた…
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は自ら前に倒れて四つん這いになると、%ANAME(MPLY:0)%に後背位で腰を振らせた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は体をひねって自ら四つん這いになると、今度は%ANAME(MPLY:0)%に後背位で腰を振らせた…
				ENDIF
				;第三者に主導権
			CASEELSE
				;前回が後背位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に後背位で挿入している%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は言われるまま腰の動きを激しくし、後ろから何度も%ANAME(MTAR:0)%の膣奥を突いた…
					;背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を後背位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を押し倒して後ろから覆い被さると、再び腰を動かし始めた…
					;正常位・対面座位・騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を後背位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を返して四つん這いにすると、再び腰を動かし始めた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
				;プレイヤーに主導権
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%は後ろから%ANAME(MTAR:0)%の%LOCALS:0%にペニスを突き入れ、何度も腰を打ち付け%TARGET_STR("た", "てきた")%…
				;ターゲットに主導権
			CASE 1
				PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に%STR_BODY("尻：外見", MTAR:0)%を向け、後ろから%LOCALS:0%にペニスを挿入させた…
				;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、後背位で%ANAME(MTAR:0)%を犯すように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を四つん這いにすると、%ANAME(MTAR:0)%の%LOCALS:0%にペニスを突き入れて腰を振った…
		ENDSELECT
	ENDIF
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST31
SIF IS_PIERCED(MPLY:0, ピアス_ペニス)
	PRINTFORML %ANAME(MPLY:0)%のペニスに取り付けられたピアスが、%ANAME(MTAR:0)%の膣内をごりごりと抉っていく……

IF NOWEX:(MTAR:0):Ｖ絶頂 >= 8
	SELECTCASE RAND:2
		CASE 0
			{
				PRINTFORML %ANAME(MTAR:0)%は%S_R("支えきれなくなった上体を投げ出し/身体を巡る激しい絶頂の波に足指をピンと張らせて/激しい絶頂の波に身体全体を仰け反らせて")%、
					%S_R("全身をびくびくと震わせて/淫らな表情で/口から唾液をこぼしながら")%%S_R("快楽を堪能して/激しく喘いで/快楽を貪って/肉欲の虜になって")%いる
			}
		CASE 1
			{
				PRINTFORML %ANAME(MTAR:0)%は%S_R("激しい絶頂の余韻を味わいながら/絶頂のあまりすっかり表情を弛緩させて/押し寄せる肉欲のうねりに飲まれて")%、
					%BAR_NAME(MPLY:0)%%S_R("を咥え込んだ尻を淫らにくねらせて/の注挿にあわせて尻を淫らにくねらせて/の与えるさらなる悦楽を求めて尻をくねらせて")%いる…
			}
	ENDSELECT
ENDIF

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN31
RETURN コマンド_ウフフ

;--------------------
;性嗜好用関数
;--------------------
@COM_PREFERENCE_PLAYER_31()
RETURN POWER(2, 性的嗜好_性交したい)

@COM_PREFERENCE_TARGET_31()
RETURN POWER(2, 性的嗜好_性交されたい)

@COM_STACK_SPERM_MPLY_TO_MTAR_31()
RETURN 射精部位_膣内


code:
;スキンシップ

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME320
;温泉なら裸のスキンシップ
IF TFLAG:デート場所 == 7
	RESULTS:0 = Get skinship(裸)
	RESULTS:1 = You can get a skinship(naked)
	RESULTS:2 = To ask for a skinship(naked)
	RESULTS:3 = Skinship (naked)
ELSE
	RESULTS:0 = Get a skinship
	RESULTS:1 = You can get a skinship
	RESULTS:2 = To ask for a skinship
	RESULTS:3 = Skinship
ENDIF
IF CFLAG:(MPLY:0):スケベモブ && (TFLAG:リアクションコマンド実行 || TFLAG:割込コマンド実行)
	RESULTS:0 '= TR("撫で回される")
	RESULTS:1 '= TR("撫で回される")
	RESULTS:2 '= TR("撫で回される")
	RESULTS:3 '= TR("撫で回される")
	RESULTS:4 '= TR("撫で回される")
	RESULTS:5 '= TR("撫で回される")
ENDIF

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE320
;共通部分
CALL COM_ABLE_COMMON(320)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
IF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	SIF !CFLAG:(MPLY:0):スケベモブ
		RETURN 0
ENDIF
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM320
;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

SIF CFLAG:(MPLY:0):スケベモブ && (TFLAG:リアクションコマンド実行 || TFLAG:割込コマンド実行)
	CALL ADD_SITU("セクハラ")

;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 100, 100)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 100, 100)

;温泉なら裸でのスキンシップ
IF TFLAG:デート場所 == 7 || (IS_BATHING(MPLY) && IS_BATHING(MTAR))
	TFLAG:17 = 2

	SOURCE:(MPLY:0):接触 = 450
	SOURCE:(MTAR:0):接触 = 450
	SOURCE:(MPLY:0):露出 = 50
	SOURCE:(MTAR:0):露出 = 50

	;親密に応じた歓楽のソース追加
	SOURCE:(MPLY:0):歓楽 = -30
	SOURCE:(MTAR:0):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MPLY:0, 150)
	CALL ADD_SOURCE_KANRAKU(MTAR:0, 150)
	SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
	SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MPLY:0, 130)
	CALL ADD_SOURCE_AIZYOU(MTAR:0, 130)

	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 30, 30)
	CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 30, 30)

	;レズ・ＢＬ経験基準値
	TFLAG:レズ・ＢＬ経験基準値 = 2

ELSE
	;恋人or烙印で濃厚なスキンシップ
	IF TALENT:(LOCAL:1):恋人 || TALENT:(LOCAL:1):烙印
		TFLAG:17 = 1

		;レズ・ＢＬ経験基準値
		TFLAG:レズ・ＢＬ経験基準値 = 1
	ELSE
		TFLAG:17 = 0
	ENDIF

	SOURCE:(MPLY:0):接触 = 150
	SOURCE:(MTAR:0):接触 = 150

	;親密に応じた歓楽のソース追加
	SOURCE:(MPLY:0):歓楽 = -30
	SOURCE:(MTAR:0):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MPLY:0, 140)
	CALL ADD_SOURCE_KANRAKU(MTAR:0, 140)
	SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
	SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MPLY:0, 90)
	CALL ADD_SOURCE_AIZYOU(MTAR:0, 90)

	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 20, 20)
	CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 20, 20)
ENDIF

IF AT_SITU("セクハラ")
	SOURCE:(MPLY:0):性行動 = 65
	SOURCE:(MTAR:0):性行動 = 65
	SOURCE:(MTAR:0):快Ｃ = SENSE_HOUSHI(MPLY:0, MTAR:0, 150)
	SOURCE:(MTAR:0):快Ｂ = SENSE_HOUSHI(MPLY:0, MTAR:0, 150)
ENDIF
;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.10
	TIMES SOURCE:(MPLY:0):優越, 0.50
	TIMES SOURCE:(MPLY:0):愛情, 0.50
	TIMES SOURCE:(MTAR:0):歓楽, 0.10
	TIMES SOURCE:(MTAR:0):優越, 0.50
	TIMES SOURCE:(MTAR:0):愛情, 0.50
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
	TCVAR:(MTAR:0):HOムード値 -= 5
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	TCVAR:(MTAR:0):HOムード値 += 2
	;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):優越, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):優越, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
	TCVAR:(MTAR:0):HOムード値 += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

CALL FL_PLAYSE(S_R("Pat_2.mp3"))

RETURN 1

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER320(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 35

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 2)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 5)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 4)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE320
;キス中
IF IS_EQUIP(MPLY:0, 342)
	LOCALS:0 = キスをしながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

;温泉でデート中の場合OR入浴コマンド実行中
IF TFLAG:デート場所 == 7 || (IS_BATHING(MPLY) && IS_BATHING(MTAR))
	;プレイヤーが主人公側
	IF MPLY:0 == MASTER
		;主導権が主人公側
		IF IS_INITIATIVE(MASTER)
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%と裸の状態で肌を寄せ合った
			;主導権が相手側
		ELSE
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、%LOCALS:0%裸の状態で肌を寄せ合った
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MTAR:0)%は何だか嬉しそうにしている…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML だが、%ANAME(MTAR:0)%に突き放されてしまった…
		ENDIF
		;プレイヤーが相手側
	ELSE
		;主導権が主人公側
		IF IS_INITIATIVE(MASTER)
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、%LOCALS:0%裸の状態で肌を寄せ合った
			;主導権が相手側
		ELSE
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の腕を取って、裸の状態で肌を寄せてきた
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応が淡白だったせいで拗ねてしまったようだ…
		ENDIF
	ENDIF

ELSE
	IF TFLAG:17 == 1
		LOCALS:1 = 少し濃厚な
	ENDIF

	;プレイヤーが主人公側
	IF MPLY:0 == MASTER
		;主導権が主人公側
		IF IS_INITIATIVE(MASTER)
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%との%LOCALS:1%スキンシップを図った
			;主導権が相手側
		ELSE
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の求めに応じて%LOCALS:1%スキンシップを図った
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MTAR:0)%は何だか嬉しそうにしている…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML どうやら、少し馴れ馴れしくしすぎたようだ…
		ENDIF
		;スケベモブ -> リアクションコマンド
	ELSEIF AT_SITU("セクハラ")
		LOCALS '= @"%ANAME(MPLY:0)%は"
		LOCALS += @"%S_R(@"下卑た笑みを浮かべて/いやらしい手つきで/荒らげた熱い吐息を浴びせながら/必要以上に身体を擦り寄せて")%"
		LOCALS += @"%ANAME(MTAR:0)%の%STR_BODY("身体", MTAR:0)%"
		LOCALS += @"%S_R("に手を這わせた/を執拗に撫で回した/のあちこちを弄った")%"
		PRINTFORMW %LOCALS%
		;プレイヤーが相手側
	ELSE
		;主導権が主人公側
		IF IS_INITIATIVE(MASTER)
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、%LOCALS:0%%LOCALS:1%スキンシップを図った
			;主導権が相手側
		ELSE
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の体を寄せ、%LOCALS:1%スキンシップを図ってきた
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応が淡白だったせいで拗ねてしまったようだ…
		ENDIF
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST320
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN320
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;頭を撫でる

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME321
RESULTS:0 = Headpat
RESULTS:1 = Be able to headpat
RESULTS:2 = Get headpat
RESULTS:3 = Headpat yourself

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE321
;共通部分
CALL COM_ABLE_COMMON(321)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM321
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	LOCAL:0 = 0
	IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		LOCAL:0 = 5
	ENDIF
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5 + LOCAL:0, 10 - LOCAL:0)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 250
SOURCE:(MTAR:0):接触 = 250

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 100)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 100)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 130)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 130)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 150, 150)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 150, 150)

IF !TALENT:(MTAR:0):幼児 && !TALENT:(MTAR:0):幼稚 && !TALENT:(MTAR:0):プライド低い
	IF TALENT:(MTAR:0):プライド高い
		SOURCE:(MTAR:0):反感 = 150
		TIMES SOURCE:(MTAR:0):歓楽, 0.50
	ELSE
		SOURCE:(MTAR:0):反感 = 60
		TIMES SOURCE:(MTAR:0):歓楽, 0.80
	ENDIF
ENDIF

IF TALENT:(MPLY:0):母性
	TIMES SOURCE:(MTAR:0):愛情, 1.50
ENDIF

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):反感, 2.00
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TCVAR:(MTAR:0):HOムード値 -= 5
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	TCVAR:(MTAR:0):HOムード値 += 2
	;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):反感, 0.50
	TCVAR:(MTAR:0):HOムード値 += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER321(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 30

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 5)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 6)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 4)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE321
#DIMS DYNAMIC 実行名
#DIMS DYNAMIC 対象名
#DIMS DYNAMIC 対象_状態
IF IS_EQUIP_PLAYER(MTAR:0, 11, 107)
	対象_状態 '= "対象_口淫"
ELSEIF IS_EQUIP_TARGET(MPLY:0, T_N("膝枕する"))
	対象_状態 '= "対象_膝枕"
ELSEIF IS_EQUIP_PLAYER(MPLY:0, T_N("膝枕する"))
	対象_状態 '= "実行_膝枕"
ENDIF
;キス中
IF IS_EQUIP(MPLY:0, 342)
	LOCALS:0 = キスをしながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

実行名 = %ANAME(MPLY:0)%
対象名 = %ANAME(MTAR:0)%

DEBUGPRINTFORML 対象_状態 = %対象_状態%

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		SELECTCASE 対象_状態
			CASE "対象_口淫"
				PRINTFORM %実行名%は
				PRINTFORM %S_R("一心不乱に/懸命に/ひたすら/",1)%
				PRINTFORM %S_R("ペニス/剛直/亀頭/淫棒",1)%
				PRINTFORM %S_R("に舌を這わせて/を舐め取って/に舌を絡めて/を舐め回して",1)%
				PRINTFORMW 奉仕する%対象名%の頭を撫でた
				PRINTL
			CASE "実行_膝枕"
				{
					CALLF PRINT_RANDC(
						TEXT_CONCATEC(
						@"%実行名%は", 1
						,@"膝の上に頭を預ける%対象名%の頭をそっと撫でた", 1
						), 1
						,TEXT_CONCATEC(
						@"%実行名%は", 1
						,@"膝の上に頭を乗せる%対象名%の髪をそっと撫でた", 1
						), 1
						)
				}
				PRINTL
			CASEELSE
				PRINTFORM %実行名%は%LOCALS:0%%対象名%
				IF CFLAG:(MTAR:0):好感度 >= 800
					PRINTFORMW を抱き寄せて優しく頭を撫でた
				ELSE
					PRINTFORMW の頭を撫でた
				ENDIF
		ENDSELECT
		;主導権が相手側
	ELSE
		PRINTFORM %対象名%は%LOCALS:0%%実行名%
		IF CFLAG:(MTAR:0):好感度 >= 800
			PRINTFORMW に体を預け、頭を撫でるように要求した
		ELSE
			PRINTFORMW に、頭を撫でて欲しいとねだった
		ENDIF
	ENDIF

	IF TFLAG:コマンド成否 == 1
		IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
			PRINTFORML %対象名%は眼を細め、心地良さそうにしている…
		ELSE
			PRINTFORML %対象名%は照れ臭そうに、笑みを浮かべている…
		ENDIF
	ELSEIF TFLAG:コマンド成否 == 0
		IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
			PRINTFORML %ANAME(MTAR:0)%は嬉しそうにしている…
		ELSE
			PRINTFORML %ANAME(MTAR:0)%は頬を紅潮させ、恥ずかしそうにしている…
		ENDIF
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML %ANAME(MTAR:0)%は子供扱いされて不機嫌そうだ…
	ENDIF
	;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORM %ANAME(MTAR:0)%は%LOCALS:0%%ANAME(MPLY:0)%に
		IF CFLAG:(MPLY:0):好感度 >= 800
			PRINTFORMW 抱きついて、甘えながら頭を撫でてもらった
		ELSE
			PRINTFORMW 頭を撫でてもらった
		ENDIF

		IF TFLAG:コマンド成否 == 1
			IF TALENT:(MPLY:0):母性
				PRINTFORML %ANAME(MPLY:0)%は我が子を愛おしむかの様に、撫でてくれた…
			ELSE
				PRINTFORML %ANAME(MPLY:0)%は笑顔で優しく撫でてくれた…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == 0
			IF TALENT:(MPLY:0):母性
				PRINTFORML %ANAME(MPLY:0)%は笑顔で優しく、ただ撫でてくれた…
			ELSE
				PRINTFORML %ANAME(MPLY:0)%は呆れた様に苦笑しながらも、優しく撫でてくれた…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML %ANAME(MPLY:0)%は心底呆れたような表情でため息をついた…
		ENDIF
		;主導権が相手側
	ELSE
		PRINTFORM %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%
		IF CFLAG:(MPLY:0):好感度 >= 800
			PRINTFORMW を抱き寄せると、甘やかすように優しく頭を撫でた
		ELSE
			PRINTFORMW の頭を撫でてきた
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
		ELSEIF TFLAG:コマンド成否 == 0
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応を見て楽しんでいる…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
		ENDIF
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST321
#DIMS DYNAMIC 対象_状態
FOR LOCAL, 0, MTAR_NUM
	IF IS_SERVEAT(MTAR:LOCAL, "口", "フェラ") && CHECK_EMOTION(MTAR:LOCAL, "GOOD")
		CALLF TAG_PRINT(@"@C:{カラー_ピンク}@%ONOMATOPE_PRESET("フェラ音長", (RAND:2), 0)%@/C@", 1)
		PRINTL
		;あまりピンクを多用すると読みにくい気がする。
		CALLF TAG_PRINT(@"@C:{カラー_白}@%ANAME(MTAR:LOCAL)%は%ANAME(MPLY:0)%%S_R("に奉仕を褒められ/に頭を撫でられながら/の手に頭を預けて")%、%S_R("嬉しそうに/頬を紅潮させ/うっとりした様子で/表情を弛緩させ/熱い吐息をもらしながら")%%S_R("一層夢中に/より丹念に/さらに積極的に")%%語彙_ペニス()%%S_R("を舐め回して/に舌を這わせて/にむしゃぶりついて/に奉仕して")%いる@C@", 1)
	ENDIF
NEXT
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------
;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN321
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;髪を梳く

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME322
RESULTS:0 = 髪を梳く
RESULTS:1 = 髪梳きさせられる
RESULTS:2 = 髪を梳いてもらう
RESULTS:3 = 髪梳きされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE322
;共通部分
CALL COM_ABLE_COMMON(322)
SIF RESULT == 0
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM322
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 250
SOURCE:(MTAR:0):接触 = 250

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 120)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 120)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 120)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 120)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 70, 70)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 70, 70)

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 1

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0

;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER322(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 30

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 3)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 3)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE322
;キス中
IF IS_EQUIP(MPLY:0, 342)
	LOCALS:0 = キスをしながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の髪に手櫛を入れた
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%に頼まれて、手櫛で髪を梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MTAR:0)%は目を閉じて気持ちよさそうにしている…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%にされるまま身を任せている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML 途中で%ANAME(MTAR:0)%の髪を引っ掛けてしまった…
		PRINTFORM %ANAME(MTAR:0)%は目尻に涙を浮かべて
		IF TALENT:(MTAR:0):恥じらい || TALENT:(MTAR:0):臆病 || TALENT:(MTAR:0):大人しい
			PRINTFORML 痛そうに頭を抑えた…
		ELSE
			PRINTFORML %ANAME(MPLY:0)%を睨んでいる…
		ENDIF
	ENDIF
;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に頼んで、%LOCALS:0%手櫛で髪を梳いてもらった
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の髪に手櫛を入れた
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MPLY:0)%は丁寧に%ANAME(MTAR:0)%の髪に指を入れている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST322
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN322
RETURN コマンド_日常


code:
;櫛で梳かす

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME323
RESULTS:0 = 櫛で髪を梳かす
RESULTS:1 = 櫛で髪を整えさせられる
RESULTS:2 = 櫛で髪を梳かしてもらう
RESULTS:3 = 櫛で髪を梳かされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE323
;共通部分
CALL COM_ABLE_COMMON(323)
SIF RESULT == 0
	RETURN 0
SIF !ITEM:櫛
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM323
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 50
SOURCE:(MTAR:0):接触 = 50

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 140)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 140)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 120)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 120)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 70, 70)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 70, 70)

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0

;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER323(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 25

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 3)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 3)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE323
;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の髪を櫛で梳かした
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に頼まれて、櫛で髪を梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MTAR:0)%は目を閉じて気持ちよさそうにしている…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%にされるまま身を任せている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML 途中で%ANAME(MTAR:0)%の髪を引っ掛けてしまった…
		PRINTFORM %ANAME(MTAR:0)%は目尻に涙を浮かべて
		IF TALENT:(MTAR:0):恥じらい || TALENT:(MTAR:0):臆病 || TALENT:(MTAR:0):大人しい
			PRINTFORML 痛そうに頭を抑えた…
		ELSE
			PRINTFORML %ANAME(MPLY:0)%を睨んでいる…
		ENDIF
	ENDIF
;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に頼んで、櫛で髪を梳いてもらった
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の髪を櫛で梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MPLY:0)%は丁寧に%ANAME(MTAR:0)%の髪を整えている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST323
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN323
RETURN コマンド_日常